17

SAS® Macro

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

INTRODUCTION

This chapter presents the 17 SAS® macros used in the previous chapters. These are also available from *www.pisa.oecd.org*. Table 17.1 presents a summary of the 17 SAS® macros. The file names are in blue and the macro names as well as their arguments are in black.

STRUCTURE OF THE SAS® MACROS

All SAS® macros have the same structure.

- The first step consists of:
  - Saving the INFILE data file into a temporary data file denoted BRRDATA and dropping all variables that are not necessary for the analysis; and
  - Sorting that file by all variables specified in the BYVAR argument.
- The second step is the iterative part of the macro:
  - The requested statistics is computed 81 times for the computation without plausible values or 405 times for the computation with plausible values;
  - At each run, the results are stored in a temporary file, with the number of the replicate and the number of the plausible values if needed then added to that file; and
  - That file is merged with another temporary file that keeps the results of all runs.
- The third step consists of data file transformations to allow the computation of the final estimate and its respective standard error. This step is quite specific to each SAS® macro.
- The fourth step is devoted to the computation of the final statistic and its respective standard error:
  - The final estimates and 80 replicates are separated, *i.e.* BRR_TEMP1 is divided into BRR_TEMP2 (final estimates) and BRR_TEMP3 (replicate estimates). In the case of the plausible values, the five estimates are averaged and the measurement error is computed;
  - The two files are then merged to create BRR_TEMP4 and the squared differences between the final estimate and the 80 replicates are computed;
  - Through a PROC UNIVARIATE procedure, the sum of the squared differences is computed and divided by 20. These results are saved in BRR_TEMP5; and
  - BRR_TEMP3 (finale estimates and, in the case of the plausible values, the measurement variance) and BRR_TEMP5 (the sampling variance estimates) are combined. The final estimates and their respective standard errors are saved in the OUTFILE datafile.
- The final step flags the statistics for not reaching minimum sample size requirements that researchers set. This step is not included in some macros.
  - The unweighted numbers of students, the unweighted number of schools and the weighted percentage of the population are firstly computed per population estimates returned by the macro;
  - The unweighted number of schools and the weighted percentage of the population are then compared with the benchmarks provided by the researchers; and
  - If these numbers are below the benchmarks, the results are flagged.

The SAS® syntax is presented hereafter.

Table 17.1
Synthesis of the 17 SAS® macros

Requested statistics	Without plausible values	With plausible values
Mean, STD, VAR, Quartiles, Median, Percentiles	**PROC_MEANS_NO_PV.sas** `%BRR_PROCMEAN(INFILE= , REPLI_ROOT= , BYVAR= , VAR= , STAT= , LIMIT= , LIMIT_CRITERIA= , ID_SCHOOL= , OUTFILE= );`	**PROC_MEANS_PV.sas** `%BRR_PROCMEAN_PV(INFILE= , REPLI_ROOT= , BYVAR= , PV_ROOT= , STAT= , LIMIT= , LIMIT_CRITERIA= , ID_SCHOOL= , OUTFILE= );`
Frequencies	**PROC_FREQ_NO_PV.sas** `%BRR_FREQ(INFILE= , REPLI_ROOT= , BYVAR= , VAR= , LIMIT= , LIMIT_CRITERIA= , ID_SCHOOL= , OUTFILE= );`	**PROC_FREQ_PV.sas** `%BRR_FREQ_PV(INFILE= , REPLI_ROOT= , BYVAR= , PV_ROOT= , LIMIT= , LIMIT_CRITERIA= , ID_SCHOOL= , OUTFILE= );`
Regression	**PROC_REG_NO_PV.sas** `%BRR_REG(INFILE= , REPLI_ROOT= , VARDEP= , EXPLICA= , BYVAR= , LIMIT= , LIMIT_CRITERIA= , ID_SCHOOL= , OUTFILE= );`	**PROC_REG_PV.sas** `%BRR_REG_PV(INFILE= , REPLI_ROOT= , EXPLICA= , BYVAR= , PV_ROOT= , LIMIT= , LIMIT_CRITERIA= , ID_SCHOOL= , OUTFILE= );`
Correlation coefficients	**PROC_CORR_NO_PV.sas** `%BRR_CORR(INFILE= , REPLI_ROOT= , BYVAR= , VAR1= , VAR2= , LIMIT= , LIMIT_CRITERIA= , ID_SCHOOL= , OUTFILE= );`	**PROC_CORR_PV.sas** `%BRR_CORR_PV(INFILE= , REPLI_ROOT= , BYVAR= , EXPLICA= , PV_ROOT= , LIMIT= , LIMIT_CRITERIA= , ID_SCHOOL= , OUTFILE= );`
Differences on mean, STD, VAR, Quartiles, Median, Percentiles	**PROC_DIF_NO_PV.sas** `%BRR_PROCMEAN_DIF(INFILE= , REPLI_ROOT= , BYVAR= , VAR= , COMPARE= , CATEGORY= , STAT= , OUTFILE= );`	**PROC_DIF_PV.sas** `%BRR_PROCMEAN_DIF_PV(INFILE= , REPLI_ROOT= , BYVAR= , PV_ROOT= , COMPARE= , CATEGORY= , STAT= , OUTFILE= );`
Quartiles		**QUARTILE _PV.sas** `%QUARTILE_PV(INFILE= , REPLI_ROOT= , BYVAR= , PV_ROOT= , INDEX= , LIMIT= , LIMIT_CRITERIA= , ID_SCHOOL= , OUTFILE= );`
Relative risk	**RELATIVE_RISK_NO_PV.sas** `%BRR_RR(INFILE= , REPLI_ROOT= , BYVAR= , ANTECEDENT= , OUTCOME= , LIMIT= , LIMIT_CRITERIA= , ID_SCHOOL= , OUTFILE= );`	**RELATIVE_RISK_PV.sas** `%BRR_RR_PV(INFILE= , REPLI_ROOT= , BYVAR= , ANTECEDENT_ROOT= , OUTCOME_ROOT= , LIMIT= , LIMIT_CRITERIA= , ID_SCHOOL= , OUTFILE= );`
Effect size	**EFFECT_SIZE_NO_PV.sas** `%BRR_EFFECT(INFILE= , REPLI_ROOT= , BYVAR= , VAR= , EFFECT= , OUTFILE= );`	**EFFECT_SIZE_PV.sas** `%BRR_EFFECT_PV(INFILE= , REPLI_ROOT= , BYVAR= , PV_ROOT= , EFFECT= , OUTFILE= );`
Multilevel regression	**PROC_MIXED_NO_PV.sas** `%BRR_MIXED(INFILE= , REPLI_ROOT= , VARDEP= , FIXEF= , RANEF= , BYVAR= , LEVEL2= , OUTSCREEN= , OUTFILE= );`	**PROC_MIXED_PV.sas** `%BRR_MIXED_PV(INFILE= , REPLI_ROOT= , PV_ROOT= , FIXEF= , RANEF= , BYVAR= , LEVEL2= , OUTSCREEN= , OUTFILE= );`

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 17.1 [1/3] **SAS® macro of PROC_MEANS_NO_PV.sas**

```
%MACRO BRR_PROCMEAN(INFILE =,
                    REPLI_ROOT =,
                    BYVAR = ,
                    VAR =,
                    STAT =,
                    LIMIT=,
                    LIMIT_CRITERIA=,
                    ID_SCHOOL=,
                    OUTFILE =);

/*

MEANING OF THE MACRO ARGUMENTS

INFILE   = INPUT DATA FILE.
REPLI_ROOT = ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL WEIGHT
VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
BYVAR = BREAKDOWN VARIABLES.
VAR = VARIABLE ON WHICH THE REQUESTED STATISTIC WILL BE COMPUTED.
STAT =   REQUESTED STATISTIQUE.
         SUMWGT = SUM OF THE WEIGHT
         MEAN   = MEAN
         VAR =  VARIANCE
         STD = STANDARD DEVIATION
         CV =   COEFFICIENT OF VARIATION
         MEDIAN =      MEDIAN
         Q1 =   FIRST QUARTILE
         Q3 =   THIRD QUARTILE
         QRANGE = RANGE BETWEEN Q1 AND Q3
         PX =   PERCENTILE, WITH P1, P5, P10, P25, P50, P75, P90, P95 and P99
LIMIT = FLAGGING  YES OR NO.
LIMIT_CRITERIA = 1) NUMBER OF STUDENTS 2) NUMBER OF SCHOOLS 3) PERCENTAGE OF STUDENTS
AND 4) NUMBER OF VARIABLES FROM THE BYVAR ARGUMENT FOR DEFINING THE POPULATION OF
REFERENCE.
ID_SCHOOL    =   VARIABLE NAME FOR THE SCHOOL IDENTIFICATION.
OUTFILE =    FILE WITH THE STATISTIC ESTIMATES AND THEIR STANDARD ERROR ESTIMATES.

*/

OPTIONS NONOTES;

PROC DATASETS LIBRARY=WORK NOLIST;
         DELETE BRR_TEMP1;
RUN;

PROC  SORT  DATA=&INFILE  OUT=BRRDATA(KEEP=&REPLI_ROOT.0-&REPLI_ROOT.80  &BYVAR  &VAR
&ID_SCHOOL);
  BY &BYVAR;
RUN;

%DO I = 0 %TO 80;
         PROC MEANS DATA=BRRDATA VARDEF=WGT NOPRINT;
             VAR &VAR ;
             BY &BYVAR;
             WEIGHT &REPLI_ROOT&I;
             OUTPUT OUT=MEAN_TEMP &STAT=PV;
         RUN;

         DATA MEAN_TEMP;
             SET MEAN_TEMP;
             L=&I;
         RUN;

         PROC APPEND BASE = BRR_TEMP1 DATA=MEAN_TEMP;
         RUN;
%END;
```

Box 17.1 [2/3] **SAS® macro of PROC_MEANS_NO_PV.sas**

```
DATA BRR_TEMP4;
         MERGE BRR_TEMP2 BRR_TEMP3;
         BY &BYVAR;
         VARI=((PV-STAT)**2)*(1/20);
RUN;

PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
         VAR VARI;
         BY &BYVAR;
         OUTPUT OUT=BRR_TEMP5 SUM=SS;
RUN;


DATA BRR_TEMP6;
         MERGE BRR_TEMP3 BRR_TEMP5;
         BY &BYVAR;
         SESTAT=(SS)**0.5;
         FORMAT STAT F10.2;
         FORMAT SESTAT F10.2;
         KEEP &BYVAR STAT SESTAT;
RUN;

%IF (%UPCASE(&LIMIT)=NO) %THEN %DO;

         DATA &OUTFILE;
            SET BRR_TEMP6;
         RUN;

%END;
%ELSE %DO;
         DATA BRR_TEMP7;
            SET BRRDATA;
            NB_MISS=0;
            ARRAY LIST_VAR (1) &VAR;
            DO K=1 TO 1;
               IF (LIST_VAR(K) IN (.,.I,.M,.N)) THEN NB_MISS=NB_MISS+1;
            END;
            IF (NB_MISS>1) THEN NB_MISS=1;
         RUN;
         PROC FREQ DATA=BRR_TEMP7 NOPRINT;
            TABLE NB_MISS /OUT=BRR_TEMP8;
            BY &BYVAR;
            WHERE (NB_MISS=0);
         RUN;
         %LET FLAG_STUD=%SCAN(&LIMIT_CRITERIA,1);
         DATA BRR_TEMP8;
            SET BRR_TEMP8;
            IF (COUNT < &FLAG_STUD) THEN FLAG_STUD=1;
            ELSE FLAG_STUD=0;
            KEEP &BYVAR FLAG_STUD;
         RUN;
         PROC SORT DATA=BRR_TEMP7;
            BY &BYVAR &ID_SCHOOL;
         RUN;
         PROC FREQ DATA=BRR_TEMP7 NOPRINT;
            TABLE NB_MISS /OUT=BRR_TEMP9;
            BY &BYVAR &ID_SCHOOL;
            WHERE (NB_MISS=0);
         RUN;
         PROC FREQ DATA=BRR_TEMP9 NOPRINT;
            TABLE NB_MISS /OUT=BRR_TEMP10;
            BY &BYVAR;
         RUN;
```

Box 17.1 [3/3] **SAS® macro of PROC_MEANS_NO_PV.sas**

```
%LET FLAG_SCH=%SCAN(&LIMIT_CRITERIA,2);
         DATA BRR_TEMP10;
             SET BRR_TEMP10;
             IF (COUNT < &FLAG_SCH) THEN FLAG_SCH=1;
             ELSE FLAG_SCH=0;
             KEEP &BYVAR FLAG_SCH;
         RUN;
         PROC SORT DATA=BRR_TEMP7;
             BY &BYVAR NB_MISS;
         RUN;
         PROC FREQ DATA=BRR_TEMP7 NOPRINT;
             TABLE NB_MISS/OUT=BRR_TEMP11;
             BY &BYVAR;
             WEIGHT &REPLI_ROOT.0;
         RUN;
         %LET K=1;
         %LET POPREF=;
         %LET NBVAR=%SCAN(&LIMIT_CRITERIA,4);
         %DO %WHILE(&K <= &NBVAR);
             %LET POPREF_ADD=%SCAN(&BYVAR,&K);
             %LET POPREF=&POPREF &POPREF_ADD;
             %LET K=%EVAL(&K+1);
         %END;

         PROC MEANS DATA=BRR_TEMP11 NOPRINT;
             VAR COUNT;
             BY &POPREF;
             OUTPUT OUT=BRR_TEMP12 SUM=SOMWGT;
         RUN;
         %LET FLAG_PCT=%SCAN(&LIMIT_CRITERIA,3);
         DATA BRR_TEMP13;
             MERGE BRR_TEMP11 BRR_TEMP12;
             BY &POPREF;
             PCT=(COUNT/SOMWGT)*100;
             IF (PCT < &FLAG_PCT) THEN FLAG_PCT=1;
             ELSE FLAG_PCT=0;
             IF (NB_MISS=0);
             KEEP &BYVAR FLAG_PCT;
         RUN;
         DATA &OUTFILE;
             MERGE BRR_TEMP6 BRR_TEMP8 BRR_TEMP10 BRR_TEMP13;
             BY &BYVAR;
         RUN;
         PROC DATASETS LIBRARY=WORK NOLIST;
             DELETE BRR_TEMP7 BRR_TEMP8 BRR_TEMP9 BRR_TEMP10 BRR_TEMP11 BRR_TEMP12
BRR_TEMP13;
         RUN;

%END;

PROC DATASETS LIBRARY=WORK NOLIST;
         DELETE BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5 BRR_TEMP6 MEAN_
TEMP BRRDATA;
RUN;

OPTIONS NOTES;

%MEND BRR_PROCMEAN;
```

Box 17.2 [1/3] **SAS® macro of PROC_MEANS_PV.sas**

```
%MACRO BRR_PROCMEAN_PV(INFILE =,
                       REPLI_ROOT =,
                       BYVAR = ,
                       PV_ROOT =,
                       STAT =,
                       LIMIT=,
                       LIMIT_CRITERIA=,
                       ID_SCHOOL=,
                       OUTFILE =);

/*

MEANING OF THE MACRO ARGUMENTS

INFILE  = INPUT DATA FILE.
REPLI_ROOT = ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL
WEIGHT VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
BYVAR = BREAKDOWN VARIABLES
PV_ROOT = ROOT OF THE 5 PLAUSIBLE VALUES VARIABLES NAMES
STAT =  REQUESTED STATISTIQUE.
        SUMWGT = SUM OF THE WEIGHT
        MEAN = MEAN
        VAR = VARIANCE
        STD = STANDARD DEVIATION
        CV = COEFFICIENT OF VARIATION
        MEDIAN = MEDIAN
        Q1 = FIRST QUARTILE
        Q3 = THIRD QUARTILE
        QRANGE = RANGE BETWEEN Q1 AND Q3
        PX = PERCENTILE, WITH P1, P5, P10, P25, P50, P75, P90, P95 and P99
LIMIT = FLAGGING  YES OR NO.
LIMIT_CRITERIA = 1) NUMBER OF STUDENTS 2) NUMBER OF SCHOOLS 3) PERCENTAGE OF
STUDENTS AND 4) NUMBER OF VARIABLES FROM THE BYVAR ARGUMENT FOR DEFINING THE
POPULATION OF REFERENCE.
ID_SCHOOL = VARIABLE NAME FOR THE SCHOOL IDENTIFICATION
OUTFILE = FILE WITH THE STATISTIC ESTIMATES AND THEIR STANDARD ERROR ESTIMATES.

*/

OPTIONS NONOTES;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP1;
RUN;

PROC SORT DATA=&INFILE
        OUT=BRRDATA(KEEP=&REPLI_ROOT.0-&REPLI_ROOT.80 &BYVAR &PV_ROOT.1-&PV_ROOT.5
&ID_SCHOOL);
        BY &BYVAR;
RUN;

%DO I = 0 %TO 80;
        PROC MEANS DATA=BRRDATA VARDEF=WGT NOPRINT;
        VAR &PV_ROOT.1 - &PV_ROOT.5 ;
        BY &BYVAR;
        WEIGHT &REPLI_ROOT&I;
        OUTPUT OUT=MEAN_TEMP &STAT=PV1 - PV5;
        RUN;

        DATA MEAN_TEMP;
                SET MEAN_TEMP;
                L=&I;
        RUN;

        PROC APPEND BASE = BRR_TEMP1 DATA=MEAN_TEMP;
        RUN;

%END;
```

Box 17.2 [2/3] **SAS® macro of PROC_MEANS_PV.sas**

```
DATA BRR_TEMP2(DROP=STAT FIN1-FIN5 MESVAR) BRR_TEMP3(KEEP=&BYVAR STAT FIN1-FIN5
MESVAR);
        SET BRR_TEMP1;
    IF L > 0 THEN OUTPUT BRR_TEMP2;
    ELSE DO;
        STAT =(PV1+PV2+PV3+PV4+PV5)/5;
                FIN1=PV1;
                FIN2=PV2;
                FIN3=PV3;
                FIN4=PV4;
                FIN5=PV5;
        MESVAR=(((STAT-FIN1)**2)+((STAT-FIN2)**2)+((STAT-FIN3)**2)+((STAT-
FIN4)**2)+((STAT-FIN5)**2))/4;
        OUTPUT BRR_TEMP3;
        END;
RUN;
PROC SORT DATA=BRR_TEMP2;
        BY &BYVAR;
RUN;
PROC SORT DATA=BRR_TEMP3;
        BY &BYVAR;
RUN;
DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY &BYVAR;
        ARRAY A (5)
                PV1-PV5;
        ARRAY B (5)
                FIN1-FIN5;
        ARRAY C(5)
                VAR1-VAR5;
        DO I=1 TO 5;
        C(I)=(1/20)*((A(I)-B(I))**2);
        END;
RUN;
PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
        VAR VAR1 VAR2 VAR3 VAR4 VAR5;
        BY &BYVAR;
        OUTPUT OUT=BRR_TEMP5 SUM=SS1 SS2 SS3 SS4 SS5;
RUN;
DATA BRR_TEMP6;
        MERGE BRR_TEMP3 BRR_TEMP5;
        BY &BYVAR;
        SAMP=(SS1+SS2+SS3+SS4+SS5)/5;
        FINVAR=(SAMP+(1.2*MESVAR));
        SESTAT=(FINVAR)**0.5;
        FORMAT STAT F10.1;
        FORMAT SESTAT F10.2;
        sestat2=(ss1+(1.2*MESVAR))**0.5;
        KEEP &BYVAR STAT SESTAT sestat2 ;
RUN;
%IF (%UPCASE(&LIMIT)=NO) %THEN %DO;

        DATA &OUTFILE;
                SET BRR_TEMP6;
        RUN;
%END;
%ELSE %DO;
        DATA BRR_TEMP7;
                SET BRRDATA;
                NB_MISS=0;
        RUN;
        PROC FREQ DATA=BRR_TEMP7 NOPRINT;
                TABLE NB_MISS /OUT=BRR_TEMP8;
                BY &BYVAR;
        RUN;
```

Box 17.2 [3/3] **SAS® macro of PROC_MEANS_PV.sas**

```
%LET FLAG_STUD=%SCAN(&LIMIT_CRITERIA,1);
        DATA BRR_TEMP8;
                SET BRR_TEMP8;
                IF (COUNT < &FLAG_STUD) THEN FLAG_STUD=1;
                ELSE FLAG_STUD=0;
                KEEP &BYVAR FLAG_STUD;
        RUN;
        PROC SORT DATA=BRR_TEMP7;
                BY &BYVAR &ID_SCHOOL;
        RUN;
        PROC FREQ DATA=BRR_TEMP7 NOPRINT;
                TABLE NB_MISS /OUT=BRR_TEMP9;
                BY &BYVAR &ID_SCHOOL;
        RUN;
        PROC FREQ DATA=BRR_TEMP9 NOPRINT;
                TABLE NB_MISS /OUT=BRR_TEMP10;
                BY &BYVAR;
        RUN;
        %LET FLAG_SCH=%SCAN(&LIMIT_CRITERIA,2);
        DATA BRR_TEMP10;
                SET BRR_TEMP10;
                IF (COUNT < &FLAG_SCH) THEN FLAG_SCH=1;
                ELSE FLAG_SCH=0;
                KEEP &BYVAR FLAG_SCH;
        RUN;
        PROC SORT DATA=BRR_TEMP7;
                BY &BYVAR NB_MISS;
        RUN;
        PROC FREQ DATA=BRR_TEMP7 NOPRINT;
                TABLE NB_MISS/OUT=BRR_TEMP11;
                BY &BYVAR;
                WEIGHT &REPLI_ROOT.0;
        RUN;
        %LET K=1;
        %LET POPREF=;
        %LET NBVAR=%SCAN(&LIMIT_CRITERIA,4);
        %DO %WHILE(&K <= &NBVAR);
                %LET POPREF_ADD=%SCAN(&BYVAR,&K);
                %LET POPREF=&POPREF &POPREF_ADD;
                %LET K=%EVAL(&K+1);
        %END;
        PROC MEANS DATA=BRR_TEMP11 NOPRINT;
                VAR COUNT;
                BY &POPREF;
                OUTPUT OUT=BRR_TEMP12 SUM=SOMWGT;
        RUN;
        %LET FLAG_PCT=%SCAN(&LIMIT_CRITERIA,3);
        DATA BRR_TEMP13;
                MERGE BRR_TEMP11 BRR_TEMP12;
                BY &POPREF;
                PCT=(COUNT/SOMWGT)*100;
                IF (PCT < &FLAG_PCT) THEN FLAG_PCT=1;
                ELSE FLAG_PCT=0;
                KEEP &BYVAR FLAG_PCT;
        RUN;
        DATA &OUTFILE;
                MERGE BRR_TEMP6 BRR_TEMP8 BRR_TEMP10 BRR_TEMP13;
                BY &BYVAR;
        RUN;
        PROC DATASETS LIBRARY=WORK NOLIST;
                DELETE BRR_TEMP7 BRR_TEMP8 BRR_TEMP9 BRR_TEMP10 BRR_TEMP11 BRR_
TEMP12 BRR_TEMP13;
        RUN;
%END;
PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5 BRR_TEMP6 MEAN_
TEMP BRRDATA;
RUN;
OPTIONS NOTES;
%MEND BRR_PROCMEAN_PV;
```

Box 17.3 [1/3] **SAS® macro of PROC_FREQ_NO_PV.sas**

```
%macro BRR_FREQ(INFILE=,
                REPLI_ROOT=,
                BYVAR=,
                VAR=,
                LIMIT=,
                LIMIT_CRITERIA=,
                ID_SCHOOL=,
                OUTFILE=);
/*
MEANING OF THE MACRO ARGUMENTS

INFILE  = INPUT DATA FILE.
REPLI_ROOT = ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL
WEIGHT VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
BYVAR = BREAKDOWN VARIABLES.
VAR = VARIABLES ON WHICH PERCENTAGES WILL BE COMPUTED.
LIMIT = FLAGGING  YES OR NO.
LIMIT_CRITERIA = 1) NUMBER OF STUDENTS 2) NUMBER OF SCHOOLS 3) PERCENTAGE OF
STUDENTS AND 4) NUMBER OF VARIABLES FROM THE BYVAR ARGUMENT FOR DEFINING THE
POPULATION OF REFERENCE.
ID_SCHOOL = VARIABLE NAME FOR THE SCHOOL IDENTIFICATION.
OUTFILE = FILE WITH THE STATISTIC ESTIMATES AND THEIR STANDARD ERROR ESTIMATES.
*/
OPTIONS NONOTES;
PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP0;
RUN;
PROC SORT DATA=&INFILE
        OUT=BRRDATA(KEEP=&REPLI_ROOT.0-&REPLI_ROOT.80 &BYVAR &VAR &ID_SCHOOL);
        BY &BYVAR &VAR;
RUN;
%DO I = 0 %TO 80;
        PROC FREQ DATA=BRRDATA NOPRINT;
                TABLE &VAR /OUT=FREQ_TEMP ;
                BY &BYVAR;
                WEIGHT &REPLI_ROOT&I;
        RUN;
        DATA FREQ_TEMP;
                SET FREQ_TEMP;
                L=&I;
                KEEP &BYVAR L &VAR PERCENT;
        RUN;
        PROC APPEND BASE = BRR_TEMP0 DATA=FREQ_TEMP;
        RUN;
%END;
PROC SORT DATA=BRR_TEMP0 OUT=BRR_TEMP1 (RENAME=(PERCENT=PV));
        BY &BYVAR L &VAR;
RUN;
DATA BRR_TEMP2(KEEP=&BYVAR &VAR PV)BRR_TEMP3(KEEP=&BYVAR &VAR STAT);
        SET BRR_TEMP1;
        IF L > 0 THEN OUTPUT BRR_TEMP2;
        ELSE DO;
                STAT =PV;
                OUTPUT BRR_TEMP3;
        END;
RUN;
PROC SORT DATA=BRR_TEMP2;
        BY &BYVAR &VAR;
RUN;
PROC SORT DATA=BRR_TEMP3;
        BY &BYVAR &VAR;
RUN;
```

Box 17.3 [2/3] **SAS® macro of PROC_FREQ_NO_PV.sas**

```
DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY &BYVAR &VAR;
        VARI=((PV-STAT)**2)*(1/20);
RUN;

PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
        VAR VARI;
        BY &BYVAR &VAR;
        OUTPUT OUT=BRR_TEMP5 SUM=SS;
RUN;

DATA BRR_TEMP6;
        MERGE BRR_TEMP3 BRR_TEMP5;
        BY &BYVAR &VAR;
        SESTAT=(SS)**0.5;
        FORMAT STAT F10.2;
        FORMAT SESTAT F10.2;
        KEEP &BYVAR &VAR STAT SESTAT;
RUN;

%IF (%UPCASE(&LIMIT)=NO) %THEN %DO;

        DATA &OUTFILE;
                SET BRR_TEMP6;
        RUN;

%END;
%ELSE %DO;

        PROC FREQ DATA=BRRDATA NOPRINT;
                TABLE &VAR /OUT=BRR_TEMP7;
                BY &BYVAR;
        RUN;
        %LET FLAG_STUD=%SCAN(&LIMIT_CRITERIA,1);
        DATA BRR_TEMP7;
                SET BRR_TEMP7;
                IF (COUNT < &FLAG_STUD) THEN FLAG_STUD=1;
                ELSE FLAG_STUD=0;
                KEEP &BYVAR &VAR FLAG_STUD;
        RUN;
        PROC SORT DATA=BRRDATA;
                BY &BYVAR  &ID_SCHOOL;
        RUN;
        PROC FREQ DATA=BRRDATA NOPRINT;
                TABLE &VAR /OUT=BRR_TEMP8;
                BY &BYVAR  &ID_SCHOOL;
        RUN;
        PROC FREQ DATA=BRR_TEMP8 NOPRINT;
                TABLE &VAR /OUT=BRR_TEMP9;
                BY &BYVAR;
        RUN;
        %LET FLAG_SCH=%SCAN(&LIMIT_CRITERIA,2);
        DATA BRR_TEMP9;
                SET BRR_TEMP9;
                IF (COUNT < &FLAG_SCH) THEN FLAG_SCH=1;
                ELSE FLAG_SCH=0;
                KEEP &BYVAR &VAR FLAG_SCH;
        RUN;

        PROC FREQ DATA=BRRDATA NOPRINT;
                TABLE &VAR/OUT=BRR_TEMP10;
                BY &BYVAR;
                WEIGHT &REPLI_ROOT.0;
        RUN;
```


PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 17.3 [3/3] **SAS® macro of PROC_FREQ_NO_PV.sas**

```
%LET K=1;
        %LET POPREF=;
        %LET NBVAR=%SCAN(&LIMIT_CRITERIA,4);
        %DO %WHILE(&K <= &NBVAR);
                %LET POPREF_ADD=%SCAN(&BYVAR,&K);
                %LET POPREF=&POPREF &POPREF_ADD;
                %LET K=%EVAL(&K+1);
        %END;

        PROC MEANS DATA=BRR_TEMP10 NOPRINT;
                VAR COUNT;
                BY &POPREF;
                OUTPUT OUT=BRR_TEMP11 SUM=SOMWGT;
        RUN;
        %LET FLAG_PCT=%SCAN(&LIMIT_CRITERIA,3);
        DATA BRR_TEMP12;
                MERGE BRR_TEMP10 BRR_TEMP11;
                BY &POPREF;
                PCT=(COUNT/SOMWGT)*100;
                IF (PCT < &FLAG_PCT) THEN FLAG_PCT=1;
                ELSE FLAG_PCT=0;
                KEEP &BYVAR &VAR FLAG_PCT;
        RUN;
        DATA &OUTFILE;
                MERGE BRR_TEMP6 BRR_TEMP7 BRR_TEMP9 BRR_TEMP12;
                BY &BYVAR &VAR;
                IF (STAT NE .);
        RUN;


        PROC DATASETS LIBRARY=WORK NOLIST;
                DELETE BRR_TEMP7 BRR_TEMP8 BRR_TEMP9 BRR_TEMP10 BRR_TEMP11 BRR_TEMP12;
        RUN;
%END;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP0 BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5 BRR_TEMP6
FREQ_TEMP BRRDATA;
RUN;

OPTIONS NOTES;

%MEND BRR_FREQ;
```


PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 17.4 [1/4] **SAS® macro of PROC_FREQ_PV.sas**

```
%macro BRR_FREQ_PV(INFILE=,
                   REPLI_ROOT=,
                   BYVAR=,
                   PV_ROOT=,
                   LIMIT=,
                   LIMIT_CRITERIA=,
                   ID_SCHOOL=,
                   OUTFILE=);
/*

MEANING OF THE MACRO ARGUMENTS
INFILE  = INPUT DATA FILE.
REPLI_ROOT = ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL
WEIGHT VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
BYVAR = BREAKDOWN VARIABLES.
PV_ROOT =       ROOT OF THE 5 PROFICIENCY LEVEL VARIABLES NAMES.
LIMIT = FLAGGING  YES OR NO.
LIMIT_CRITERIA = 1) NUMBER OF STUDENTS 2) NUMBER OF SCHOOLS 3) PERCENTAGE OF
STUDENTS AND 4) NUMBER OF VARIABLES FROM THE BYVAR ARGUMENT FOR DEFINING THE
POPULATION OF REFERENCE.
ID_SCHOOL       =       VARIABLE NAME FOR THE SCHOOL IDENTIFICATION.
OUTFILE =       FILE WITH THE STATISTIC ESTIMATES AND THEIR STANDARD ERROR
ESTIMATES.

*/
OPTIONS NONOTES;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP0;
RUN;
%DO I = 1 %TO 5;
        PROC SORT DATA=&INFILE
                OUT=BRRDATA(KEEP=&REPLI_ROOT.0-&REPLI_ROOT.80 &BYVAR &PV_ROOT.1-&PV_
ROOT.5 &ID_SCHOOL);
                BY &BYVAR &PV_ROOT&I;
        RUN;
        %DO J = 0 %TO 80;
                PROC FREQ DATA=BRRDATA NOPRINT;
                        TABLE &PV_ROOT&I /OUT=FREQ_TEMP ;
                        BY &BYVAR;
                        WEIGHT &REPLI_ROOT&J;
                RUN;
                DATA FREQ_TEMP;
                        SET FREQ_TEMP;
                        K=&I;
                        L=&J;
                        &PV_ROOT=&PV_ROOT&I;
                        KEEP &BYVAR K L PERCENT &PV_ROOT;
                RUN;
                PROC APPEND BASE = BRR_TEMP0 DATA=FREQ_TEMP;
                RUN;
        %END;
%END;
PROC SORT DATA=BRR_TEMP0;
        BY &BYVAR L &PV_ROOT;
RUN;
PROC TRANSPOSE DATA=BRR_TEMP0 OUT=BRR_TEMP1 PREFIX=PV;
        BY &BYVAR L &PV_ROOT;
        VAR PERCENT;
RUN;
DATA BRR_TEMP1;
        SET BRR_TEMP1;
        DROP _NAME_ _LABEL_;
RUN;
```

Box 17.4 [2/4] **SAS® macro of PROC_FREQ_PV.sas**

```
DATA BRR_TEMP2(DROP=STAT FIN1-FIN5 MESVAR)BRR_TEMP3(KEEP=&BYVAR &PV_ROOT STAT FIN1-
FIN5 MESVAR);
        SET BRR_TEMP1;
        IF L > 0 THEN OUTPUT BRR_TEMP2;
        ELSE DO;
                STAT =(PV1+PV2+PV3+PV4+PV5)/5;
                FIN1=PV1;
                FIN2=PV2;
                FIN3=PV3;
                FIN4=PV4;
                FIN5=PV5;
                MESVAR=(((STAT-FIN1)**2)+((STAT-FIN2)**2)+((STAT-FIN3)**2)+((STAT-
FIN4)**2)+((STAT-FIN5)**2))/4;
                OUTPUT BRR_TEMP3;
        END;
RUN;
PROC SORT DATA=BRR_TEMP2;
        BY &BYVAR &PV_ROOT;
RUN;
PROC SORT DATA=BRR_TEMP3;
        BY &BYVAR &PV_ROOT;
RUN;
DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY &BYVAR &PV_ROOT;
        ARRAY A (5)
                PV1-PV5;
        ARRAY B (5)
                FIN1-FIN5;
        ARRAY C(5)
                VAR1-VAR5;
        DO I=1 TO 5;
                C(I)=(1/20)*((A(I)-B(I))**2);
        END;
RUN;
PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
        VAR VAR1 VAR2 VAR3 VAR4 VAR5;
        BY &BYVAR &PV_ROOT;
        OUTPUT OUT=BRR_TEMP5 SUM=SS1 SS2 SS3 SS4 SS5;
RUN;
DATA BRR_TEMP6;
        MERGE BRR_TEMP3 BRR_TEMP5;
        BY &BYVAR &PV_ROOT;
        SAMP=(SS1+SS2+SS3+SS4+SS5)/5;
        FINVAR=(SAMP+(1.2*MESVAR));
        SESTAT=(FINVAR)**0.5;
        KEEP &BYVAR &PV_ROOT STAT SESTAT;
RUN;
%IF (%UPCASE(&LIMIT)=NO) %THEN %DO;

        DATA &OUTFILE;
                SET BRR_TEMP6;
        RUN;

%END;
%ELSE %DO;
        %DO M=1 %TO 5;
        PROC FREQ DATA=BRRDATA NOPRINT;
                TABLE &PV_ROOT&M/OUT=BRR_TEMP7;
                BY &BYVAR;
        RUN;
        DATA BRR_TEMP8;
                SET BRR_TEMP7;
                COUNT&M=COUNT;
                &PV_ROOT=&PV_ROOT&M;
                KEEP &BYVAR &PV_ROOT COUNT&M;
        RUN;
```

Box 17.4 [3/4] **SAS® macro of PROC_FREQ_PV.sas**

```
%IF (&M=1) %THEN %DO;
                DATA BRR_TEMP9;
                        SET BRR_TEMP8;
                RUN;
        %END;
        %ELSE %DO;
                DATA BRR_TEMP9;
                        MERGE BRR_TEMP9 BRR_TEMP8;
                        BY &BYVAR &PV_ROOT;
                RUN;
        %END;
        %END;
        %LET FLAG_STUD=%SCAN(&LIMIT_CRITERIA,1);
        DATA BRR_TEMP9;
                SET BRR_TEMP9;
                COUNT=(COUNT1+COUNT2+COUNT3+COUNT4+COUNT5)/5;
                IF (COUNT < &FLAG_STUD) THEN FLAG_STUD=1;
                ELSE FLAG_STUD=0;
                KEEP &BYVAR &PV_ROOT FLAG_STUD;
        RUN;
        PROC SORT DATA=BRRDATA;
                BY &BYVAR  &ID_SCHOOL;
        RUN;
        %DO M=1 %TO 5;
        PROC FREQ DATA=BRRDATA NOPRINT;
                TABLE &PV_ROOT&M /OUT=BRR_TEMP10;
                BY &BYVAR  &ID_SCHOOL;
        RUN;
        PROC FREQ DATA=BRR_TEMP10 NOPRINT;
                TABLE &PV_ROOT&M  /OUT=BRR_TEMP11;
                BY &BYVAR;
        RUN;
        DATA BRR_TEMP12;
                SET BRR_TEMP11;
                COUNT&M=COUNT;
                &PV_ROOT=&PV_ROOT&M;
                KEEP &BYVAR &PV_ROOT COUNT&M;
        RUN;
        %IF (&M=1) %THEN %DO;
                DATA BRR_TEMP13;
                        SET BRR_TEMP12;
                RUN;
        %END;
        %ELSE %DO;
                DATA BRR_TEMP13;
                        MERGE BRR_TEMP13 BRR_TEMP12;
                        BY &BYVAR &PV_ROOT;
                RUN;
        %END;
        %END;
        %LET FLAG_SCH=%SCAN(&LIMIT_CRITERIA,2);
        DATA BRR_TEMP13;
                SET BRR_TEMP13;
                COUNT=(COUNT1+COUNT2+COUNT3+COUNT4+COUNT5)/5;
                IF (COUNT < &FLAG_SCH) THEN FLAG_SCH=1;
                ELSE FLAG_SCH=0;
                KEEP &BYVAR &PV_ROOT FLAG_SCH;
        RUN;

        %DO M=1 %TO 5;
        PROC FREQ DATA=BRRDATA NOPRINT;
                TABLE &PV_ROOT&M/OUT=BRR_TEMP14;
                BY &BYVAR;
                WEIGHT &REPLI_ROOT.0;
        RUN;
        DATA BRR_TEMP15;
                SET BRR_TEMP14;
                COUNT&M=COUNT;
                &PV_ROOT=&PV_ROOT&M;
                KEEP &BYVAR &PV_ROOT COUNT&M;
        RUN;
```

Box 17.4 [4/4] **SAS® macro of PROC_FREQ_PV.sas**

```
%IF (&M=1) %THEN %DO;
        DATA BRR_TEMP16;
                SET BRR_TEMP15;
        RUN;
       %END;
       %ELSE %DO;
       DATA BRR_TEMP16;
                MERGE BRR_TEMP16 BRR_TEMP15;
                BY &BYVAR &PV_ROOT;
        RUN;
       %END;
       %END;

       %LET K=1;
       %LET POPREF=;
       %LET NBVAR=%SCAN(&LIMIT_CRITERIA,4);
       %DO %WHILE(&K <= &NBVAR);
        %LET POPREF_ADD=%SCAN(&BYVAR,&K);
        %LET POPREF=&POPREF &POPREF_ADD;
        %LET K=%EVAL(&K+1);
       %END;

       PROC MEANS DATA=BRR_TEMP16 NOPRINT;
        VAR COUNT1-COUNT5 ;
        BY &POPREF;
        OUTPUT OUT=BRR_TEMP17 SUM=SOMWGT1-SOMWGT5;
       RUN;
       %LET FLAG_PCT=%SCAN(&LIMIT_CRITERIA,3);
       DATA BRR_TEMP18;
        MERGE BRR_TEMP16 BRR_TEMP17;
        BY &POPREF;
        ARRAY B1 (5) PCT1-PCT5;
        ARRAY B2 (5) COUNT1-COUNT5;
        ARRAY B3 (5) SOMWGT1-SOMWGT5;
        DO A=1 TO 5;
                B1(A)=(B2(A)/B3(A))*100;
        END;
        PCT=(PCT1+PCT2+PCT3+PCT4+PCT5)/5;
        IF (PCT < &FLAG_PCT) THEN FLAG_PCT=1;
        ELSE FLAG_PCT=0;
        KEEP &BYVAR &PV_ROOT FLAG_PCT;
       RUN;

       DATA &OUTFILE;
        MERGE BRR_TEMP6 BRR_TEMP9 BRR_TEMP13 BRR_TEMP18;
        BY &BYVAR &PV_ROOT;
       RUN;

       PROC DATASETS LIBRARY=WORK NOLIST;
       DELETE BRR_TEMP7 BRR_TEMP8 BRR_TEMP9 BRR_TEMP10 BRR_TEMP11 BRR_TEMP12 BRR_TEMP13
                       BRR_TEMP14 BRR_TEMP15 BRR_TEMP16 BRR_TEMP17 BRR_TEMP18;
       RUN;

%END;

PROC DATASETS LIBRARY=WORK NOLIST;
      DELETE BRR_TEMP0 BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5 FREQ_TEMP
BRR_TEMP6 MEAN_TEMP BRRDATA;
RUN;

OPTIONS NOTES;

%MEND BRR_FREQ_PV;
```

Box 17.5 [1/3] **SAS® macro of PROC_REG_NO_PV.sas**

```
%macro BRR_REG(INFILE=,
               REPLI_ROOT=,
               VARDEP=,
               EXPLICA=,
               BYVAR=,
               LIMIT=,
               LIMIT_CRITERIA=,
               ID_SCHOOL=,
               OUTFILE=);
/*
MEANING OF THE MACRO ARGUMENTS

INFILE = INPUT DATA FILE.
REPLI_ROOT = ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL
WEIGHT VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
VARDEP = DEPENDENT VARIABLE
EXPLICA = LIST OF INDEPENDENT VARIABLES
BYVAR = BREAKDOWN VARIABLES
LIMIT = FLAGGING  YES OR NO
LIMIT_CRITERIA = 1) NUMBER OF STUDENTS 2) NUMBER OF SCHOOLS 3) PERCENTAGE OF
STUDENTS AND 4) NUMBER OF VARIABLES FROM THE BYVAR ARGUMENT FOR DEFINING THE
POPULATION OF REFERENCE.
ID_SCHOOL = VARIABLE NAME FOR THE SCHOOL IDENTIFICATION
OUTFILE = FILE WITH THE STATISTIC ESTIMATES AND THEIR STANDARD ERROR ESTIMATES.
*/
OPTIONS NONOTES;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP1;
RUN;
PROC SORT DATA=&INFILE OUT=BRRDATA (KEEP=&REPLI_ROOT.0-&REPLI_ROOT.80 &BYVAR
&VARDEP &EXPLICA &ID_SCHOOL);
        BY &BYVAR;
RUN;
%DO I = 0 %TO 80;

        PROC REG DATA=BRRDATA OUTEST=COEF_TEMP NOPRINT EDF;
                MODEL &VARDEP=&EXPLICA;
                WEIGHT &REPLI_ROOT&I;
                BY &BYVAR;
        RUN;
        DATA COEF_TEMP;
                SET COEF_TEMP;
                L=&I;
        RUN;
        PROC APPEND BASE = BRR_TEMP1 DATA=COEF_TEMP;
        RUN;
%END;
PROC SORT DATA=BRR_TEMP1;
        BY &BYVAR L;
RUN;
PROC TRANSPOSE DATA=BRR_TEMP1 OUT=BRR_TEMP;
        BY &BYVAR L;
        VAR INTERCEPT &EXPLICA _RSQ_;
RUN;

DATA BRR_TEMP1 ;
        SET BRR_TEMP (RENAME=(_NAME_=CLASS COL1=COEF));
RUN;

%LET I=1;
%DO %WHILE(%LENGTH(%SCAN(&EXPLICA,&I)));
        %LET I=%EVAL(&I+1);
%END;
```

Box 17.5 [2/3] **SAS® macro of PROC_REG_NO_PV.sas**

```
%LET NB=%EVAL(&I-1);
%DO J=0 %to &NB;
        %IF &J=0 %THEN %DO;
                %LET INDEP=Intercept;
        %END;
        %IF &J>0 %THEN %DO;
                %LET INDEP=%SCAN(&EXPLICA,&J);
        %END;
        DATA BRR_TEMP1;
                SET BRR_TEMP1;
                IF (UPCASE(CLASS)=UPCASE("&INDEP")) THEN ORDRE=&J;
                IF (SUBSTR(CLASS,1,5)="_RSQ_") THEN ORDRE=&I;
        RUN;
%END;
DATA BRR_TEMP2(KEEP=&BYVAR CLASS COEF) BRR_TEMP3(KEEP=&BYVAR CLASS ORDRE STAT);
        SET BRR_TEMP1;
        IF L > 0 THEN OUTPUT BRR_TEMP2;
        ELSE DO;
                STAT =COEF;
                OUTPUT BRR_TEMP3;
        END;
RUN;
PROC SORT DATA=BRR_TEMP2;
        BY &BYVAR CLASS;
RUN;
PROC SORT DATA=BRR_TEMP3;
        BY &BYVAR CLASS;
RUN;
DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY &BYVAR CLASS;
        VARI=((COEF-STAT)**2)*(1/20);
RUN;
PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
        VAR VARI;
        BY &BYVAR CLASS;
        OUTPUT OUT=BRR_TEMP5 SUM=SS;
RUN;
DATA BRR_TEMP6;
        MERGE BRR_TEMP3 BRR_TEMP5;
        BY &BYVAR CLASS;
        SESTAT=(SS)**0.5;
        FORMAT STAT F10.2;
        FORMAT SESTAT F10.2;
        KEEP &BYVAR CLASS ORDRE STAT SESTAT;
RUN;
PROC SORT DATA=BRR_TEMP6 OUT=&OUTFILE (DROP=ORDRE);
        BY &BYVAR ORDRE;
RUN;
%IF (%UPCASE(&LIMIT)=YES) %THEN %DO;

        DATA BRR_TEMP7;
                SET BRRDATA;
                NB_MISS=0;
                ARRAY LIST_VAR (&I) &EXPLICA &VARDEP;
                DO K=1 TO &I;
                        IF (LIST_VAR(K) IN (.,.I,.M,.N)) THEN NB_MISS=NB_MISS+1;
                END;
                IF (NB_MISS>1) THEN NB_MISS=1;
        RUN;
        PROC FREQ DATA=BRR_TEMP7 NOPRINT;
                TABLE NB_MISS /OUT=BRR_TEMP8;
                BY &BYVAR;
                WHERE (NB_MISS=0);
        RUN;
```

Box 17.5 [3/3] **SAS® macro of PROC_REG_NO_PV.sas**

```
%LET FLAG_STUD=%SCAN(&LIMIT_CRITERIA,1);
        DATA BRR_TEMP8;
               SET BRR_TEMP8;
               IF (COUNT < &FLAG_STUD) THEN FLAG_STUD=1;
               ELSE FLAG_STUD=0;
               KEEP &BYVAR FLAG_STUD;
        RUN;
        PROC SORT DATA=BRR_TEMP7;
               BY &BYVAR &ID_SCHOOL;
        RUN;
        PROC FREQ DATA=BRR_TEMP7 NOPRINT;
               TABLE NB_MISS /OUT=BRR_TEMP9;
               BY &BYVAR &ID_SCHOOL;
               WHERE (NB_MISS=0);
        RUN;
        PROC FREQ DATA=BRR_TEMP9 NOPRINT;
               TABLE NB_MISS /OUT=BRR_TEMP10;
               BY &BYVAR;
        RUN;
        %LET FLAG_SCH=%SCAN(&LIMIT_CRITERIA,2);
        DATA BRR_TEMP10;
               SET BRR_TEMP10;
               IF (COUNT < &FLAG_SCH) THEN FLAG_SCH=1;
               ELSE FLAG_SCH=0;
               KEEP &BYVAR FLAG_SCH;
        RUN;
        PROC SORT DATA=BRR_TEMP7;
               BY &BYVAR NB_MISS;
        RUN;
        PROC FREQ DATA=BRR_TEMP7 NOPRINT;
               TABLE NB_MISS/OUT=BRR_TEMP11;
               BY &BYVAR;
               WEIGHT &REPLI_ROOT.0;
        RUN;
        %LET K=1;
        %LET POPREF=;
        %LET NBVAR=%SCAN(&LIMIT_CRITERIA,4);
        %DO %WHILE(&K <= &NBVAR);
               %LET POPREF_ADD=%SCAN(&BYVAR,&K);
               %LET POPREF=&POPREF &POPREF_ADD;
               %LET K=%EVAL(&K+1);
        %END;
        PROC MEANS DATA=BRR_TEMP11 NOPRINT;
               VAR COUNT;
               BY &POPREF;
               OUTPUT OUT=BRR_TEMP12 SUM=SOMWGT;
        RUN;
        %LET FLAG_PCT=%SCAN(&LIMIT_CRITERIA,3);
        DATA BRR_TEMP13;
               MERGE BRR_TEMP11 BRR_TEMP12;
               BY &POPREF;
               PCT=(COUNT/SOMWGT)*100;
               IF (PCT < &FLAG_PCT) THEN FLAG_PCT=1;
               ELSE FLAG_PCT=0;
               IF (NB_MISS=0);
               KEEP &BYVAR FLAG_PCT;
        RUN;
        DATA &OUTFILE._CRITERIA;
               MERGE BRR_TEMP8 BRR_TEMP10 BRR_TEMP13;
               BY &BYVAR;
        RUN;
        PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP7 BRR_TEMP8 BRR_TEMP9 BRR_TEMP10 BRR_TEMP11 BRR_TEMP12 BRR_TEMP13;
        RUN;
%END;
PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5 BRR_TEMP6
COEF_TEMP BRRDATA;
RUN;
OPTIONS NOTES;
%MEND BRR_REG;
```


PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 17.6 [1/4] **SAS® macro of PROC_REG_PV.sas**

```
%macro BRR_REG_PV(INFILE=,
                  REPLI_ROOT=,
                  EXPLICA=,
                  BYVAR=,
                  PV_ROOT=,
                  LIMIT=,
                  LIMIT_CRITERIA=,
                  ID_SCHOOL=,
                  OUTFILE=);
/*
MEANING OF THE MACRO ARGUMENTS
INFILE  = INPUT DATA FILE.
REPLI_ROOT = ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL
WEIGHT VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
EXPLICA = LIST OF INDEPENDENT VARIABLES
BYVAR = BREAKDOWN VARIABLES
PV_ROOT = ROOT OF THE 5 PLAUSIBLE VALUES VARIABLES NAMES
LIMIT = FLAGGING  YES OR NO
LIMIT_CRITERIA = 1) NUMBER OF STUDENTS 2) NUMBER OF SCHOOLS 3) PERCENTAGE OF
STUDENTS AND 4) NUMBER OF VARIABLES FROM THE BYVAR ARGUMENT FOR DEFINING THE
POPULATION OF REFERENCE
ID_SCHOOL = VARIABLE NAME FOR THE SCHOOL IDENTIFICATION
OUTFILE = FILE WITH THE STATISTIC ESTIMATES AND THEIR STANDARD ERROR ESTIMATES.
*/
OPTIONS NONOTES;
PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP1;
RUN;
PROC SORT DATA=&INFILE OUT=BRRDATA (KEEP=&REPLI_ROOT.0-&REPLI_ROOT.80 &BYVAR &PV_
ROOT.1-&PV_ROOT.5 &EXPLICA &ID_SCHOOL);
        BY &BYVAR;
RUN;
%DO I = 0 %TO 80;
        PROC REG DATA=BRRDATA OUTEST=COEF_TEMP NOPRINT EDF;
                MODEL &PV_ROOT.1=&EXPLICA;
                MODEL &PV_ROOT.2=&EXPLICA;
                MODEL &PV_ROOT.3=&EXPLICA;
                MODEL &PV_ROOT.4=&EXPLICA;
                MODEL &PV_ROOT.5=&EXPLICA;
                WEIGHT &REPLI_ROOT&I;
                BY &BYVAR;
        RUN;
        DATA COEF_TEMP;
                SET COEF_TEMP;
                L=&I;
        RUN;
        PROC APPEND BASE = BRR_TEMP1 DATA=COEF_TEMP;
        RUN;

%END;
PROC SORT DATA=BRR_TEMP1;
        BY &BYVAR L;
RUN;

PROC TRANSPOSE DATA=BRR_TEMP1 OUT=BRR_TEMP PREFIX=PV;
        BY &BYVAR L;
        VAR INTERCEPT &EXPLICA _RSQ_;
RUN;
DATA BRR_TEMP1 ;
        SET BRR_TEMP (RENAME=(_NAME_=CLASS));
        DROP _LABEL_;
RUN;
```

Box 17.6 [2/4] **SAS® macro of PROC_REG_PV.sas**

```
%LET I=1;
%DO %WHILE(%LENGTH(%SCAN(&EXPLICA,&I)));
        %LET I=%EVAL(&I+1);
%END;
%let NB=%EVAL(&I-1);

%DO J=0 %to &NB;
         %IF &J=0 %THEN %DO;
        %LET INDEP=Intercept;
         %END;
         %IF &J>0 %THEN %DO;
               %LET INDEP=%SCAN(&EXPLICA,&J);
         %END;

         DATA BRR_TEMP1;
               SET BRR_TEMP1;
               IF (UPCASE(CLASS)=UPCASE("&INDEP")) THEN ORDRE=&J;
               IF (SUBSTR(CLASS,1,5)="_RSQ_") THEN ORDRE=&I;
         RUN;
%END;
DATA BRR_TEMP2(KEEP=&BYVAR CLASS ORDRE PV1-PV5)BRR_TEMP3(KEEP=&BYVAR CLASS ORDRE
FIN1-FIN5 STAT MESVAR);
         SET BRR_TEMP1;
         IF L > 0 THEN OUTPUT BRR_TEMP2;
         ELSE DO;
           STAT =(PV1+PV2+PV3+PV4+PV5)/5;
           FIN1=PV1;
           FIN2=PV2;
           FIN3=PV3;
           FIN4=PV4;
           FIN5=PV5;
           MESVAR=(((STAT-FIN1)**2)+((STAT-FIN2)**2)+((STAT-FIN3)**2)+((STAT-
FIN4)**2)+((STAT-FIN5)**2))/4;
           OUTPUT BRR_TEMP3;
         END;
RUN;
PROC SORT DATA=BRR_TEMP2;
         BY &BYVAR CLASS;
RUN;

PROC SORT DATA=BRR_TEMP3;
         BY &BYVAR CLASS;
RUN;
DATA BRR_TEMP4;
         MERGE BRR_TEMP2 BRR_TEMP3;
         BY &BYVAR CLASS;
         ARRAY A (5)
               PV1-PV5;
         ARRAY B (5)
               FIN1-FIN5;
         ARRAY C(5)
               VAR1-VAR5;
         DO I=1 TO 5;
               C(I)=(1/20)*((A(I)-B(I))**2);
         END;
RUN;
PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
         VAR VAR1 VAR2 VAR3 VAR4 VAR5;
         BY &BYVAR CLASS;
         OUTPUT OUT=BRR_TEMP5 SUM=SS1 SS2 SS3 SS4 SS5;
RUN;
DATA BRR_TEMP6;
         MERGE BRR_TEMP3 BRR_TEMP5;
         BY &BYVAR CLASS;
         SAMP=(SS1+SS2+SS3+SS4+SS5)/5;
         FINVAR=(SAMP+(1.2*MESVAR));
         SESTAT=(FINVAR)**0.5;
         FORMAT STAT F10.2;
         FORMAT SESTAT F10.2;
         KEEP &BYVAR CLASS STAT SESTAT ORDRE;
RUN;
```

Box 17.6 [3/4] **SAS® macro of PROC_REG_PV.sas**

```
PROC SORT DATA=BRR_TEMP6 OUT=&OUTFILE (DROP=ORDRE);
        BY &BYVAR ORDRE;
RUN;


%IF (%UPCASE(&LIMIT)=YES) %THEN %DO;

        %LET NB_M=%EVAL(&NB+5);

        DATA BRR_TEMP7;
           SET BRRDATA;
           NB_MISS=0;
           ARRAY LIST_VAR (&NB_M) &EXPLICA &PV_ROOT.1-&PV_ROOT.5;
           DO K=1 TO &NB;
                   IF (LIST_VAR(K) IN (.,.I,.M,.N)) THEN NB_MISS=NB_MISS+1;
           END;
           IF (NB_MISS>1) THEN NB_MISS=1;
        RUN;
        PROC FREQ DATA=BRR_TEMP7 NOPRINT;
           TABLE NB_MISS /OUT=BRR_TEMP8;
           BY &BYVAR;
           WHERE (NB_MISS=0);
        RUN;
        %LET FLAG_STUD=%SCAN(&LIMIT_CRITERIA,1);
        DATA BRR_TEMP8;
           SET BRR_TEMP8;
           IF (COUNT < &FLAG_STUD) THEN FLAG_STUD=1;
           ELSE FLAG_STUD=0;
           KEEP &BYVAR FLAG_STUD;
        RUN;
        PROC SORT DATA=BRR_TEMP7;
           BY &BYVAR &ID_SCHOOL;
        RUN;
        PROC FREQ DATA=BRR_TEMP7 NOPRINT;
           TABLE NB_MISS /OUT=BRR_TEMP9;
           BY &BYVAR &ID_SCHOOL;
           WHERE (NB_MISS=0);
        RUN;
        PROC FREQ DATA=BRR_TEMP9 NOPRINT;
           TABLE NB_MISS /OUT=BRR_TEMP10;
           BY &BYVAR;
        RUN;
        %LET FLAG_SCH=%SCAN(&LIMIT_CRITERIA,2);
        DATA BRR_TEMP10;
           SET BRR_TEMP10;
           IF (COUNT < &FLAG_SCH) THEN FLAG_SCH=1;
           ELSE FLAG_SCH=0;
           KEEP &BYVAR FLAG_SCH;
        RUN;
        PROC SORT DATA=BRR_TEMP7;
           BY &BYVAR NB_MISS;
        RUN;
        PROC FREQ DATA=BRR_TEMP7 NOPRINT;
           TABLE NB_MISS/OUT=BRR_TEMP11;
           BY &BYVAR;
           WEIGHT &REPLI_ROOT.0;
        RUN;
        %LET K=1;
        %LET POPREF=;
        %LET NBVAR=%SCAN(&LIMIT_CRITERIA,4);
        %DO %WHILE(&K <= &NBVAR);
           %LET POPREF_ADD=%SCAN(&BYVAR,&K);
           %LET POPREF=&POPREF &POPREF_ADD;
           %LET K=%EVAL(&K+1);
        %END;

        PROC MEANS DATA=BRR_TEMP11 NOPRINT;
           VAR COUNT;
           BY &POPREF;
           OUTPUT OUT=BRR_TEMP12 SUM=SOMWGT;
        RUN;
```

Box 17.6 [4/4] **SAS® macro of PROC_REG_PV.sas**

```
%LET FLAG_PCT=%SCAN(&LIMIT_CRITERIA,3);
        DATA BRR_TEMP13;
                MERGE BRR_TEMP11 BRR_TEMP12;
                BY &POPREF;
                PCT=(COUNT/SOMWGT)*100;
                IF (PCT < &FLAG_PCT) THEN FLAG_PCT=1;
                ELSE FLAG_PCT=0;
                IF (NB_MISS=0);
                KEEP &BYVAR FLAG_PCT;
        RUN;
        DATA &OUTFILE._CRITERIA;
                MERGE BRR_TEMP8 BRR_TEMP10 BRR_TEMP13;
                BY &BYVAR;
        RUN;
        PROC DATASETS LIBRARY=WORK NOLIST;
                DELETE BRR_TEMP7 BRR_TEMP8 BRR_TEMP9 BRR_TEMP10 BRR_TEMP11 BRR_
TEMP12 BRR_TEMP13;
        RUN;
%END;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5 BRR_TEMP6
COEF_TEMP BRRDATA;
RUN;

OPTIONS NOTES;

%MEND BRR_REG_PV;
```

Box 17.7 [1/3] **SAS® macro of PROC_CORR_NO_PV.sas**

```
%macro BRR_CORR(INFILE=,
                REPLI_ROOT=,
                BYVAR=,
                VAR1=,
                VAR2=,
                LIMIT=,
                LIMIT_CRITERIA=,
                ID_SCHOOL=,
                OUTFILE=);
/*
MEANING OF THE MACRO ARGUMENTS

INFILE  = INPUT DATA FILE.
REPLI_ROOT = ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL
WEIGHT VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
BYVAR = BREAKDOWN VARIABLES.
VAR1 =  FIRST NUMERIC VARIABLE.
VAR2 =  SECOND NUMERIC VARIABLE.
LIMIT = FLAGGING YES OR NO.
LIMIT_CRITERIA = 1) NUMBER OF STUDENTS 2) NUMBER OF SCHOOLS 3) PERCENTAGE OF
STUDENTS AND 4) NUMBER OF VARIABLES FROM THE BYVAR ARGUMENT FOR DEFINING THE
POPULATION OF REFERENCE.
ID_SCHOOL = VARIABLE NAME FOR THE SCHOOL IDENTIFICATION.
OUTFILE = FILE WITH THE STATISTIC ESTIMATES AND THEIR RESPECTIVE STANDARD ERROR ESTIMATE.

*/

OPTIONS NONOTES;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP1;
RUN;

PROC SORT DATA=&INFILE OUT=BRRDATA (KEEP=&REPLI_ROOT.0-&REPLI_ROOT.80 &BYVAR &VAR1
&VAR2 &ID_SCHOOL);
        BY &BYVAR;
RUN;

%DO I = 0 %TO 80;

        PROC CORR DATA=BRRDATA VARDEF=WGT NOPRINT OUTP=CORR_TEMP;
                VAR &VAR1 ;
                WITH &VAR2;
                BY &BYVAR;
                WEIGHT &REPLI_ROOT&I;
        RUN;

        DATA CORR_TEMP;
                SET CORR_TEMP;
                L=&I;
        RUN;

        PROC APPEND BASE = BRR_TEMP1 DATA=CORR_TEMP;
        RUN;
%END;

DATA BRR_TEMP1;
        SET BRR_TEMP1;
        IF (_TYPE_ NE "CORR") THEN DELETE;
        PV=&VAR1;
        KEEP &BYVAR L PV;
RUN;

DATA BRR_TEMP2(KEEP=&BYVAR PV) BRR_TEMP3(KEEP=&BYVAR STAT);
        SET BRR_TEMP1;
        IF L > 0 THEN OUTPUT BRR_TEMP2;
        ELSE DO;
                STAT =PV;
                OUTPUT BRR_TEMP3;
        END;
RUN;
```

Box 17.7 [2/3] **SAS® macro of PROC_CORR_NO_PV.sas**

```
PROC SORT DATA=BRR_TEMP2;
        BY &BYVAR;
RUN;

PROC SORT DATA=BRR_TEMP3;
        BY &BYVAR;
RUN;

DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY &BYVAR;
        VARI=((PV-STAT)**2)*(1/20);
RUN;

PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
        VAR VARI;
        BY &BYVAR;
        OUTPUT OUT=BRR_TEMP5 SUM=SS;
RUN;

DATA BRR_TEMP6;
        MERGE BRR_TEMP3 BRR_TEMP5;
        BY &BYVAR;
        SESTAT=(SS)**0.5;
        FORMAT STAT F5.2;
        FORMAT SESTAT F5.2;
        KEEP &BYVAR STAT SESTAT;
RUN;


%IF (%UPCASE(&LIMIT)=NO) %THEN %DO;

        DATA &OUTFILE;
                SET BRR_TEMP6;
        RUN;

%END;
%ELSE %DO;
        DATA BRR_TEMP7;
                SET BRRDATA;
                NB_MISS=0;
                ARRAY LIST_VAR (2) &VAR1 &VAR2;
                DO K=1 TO 2;
                        IF (LIST_VAR(K) IN (.,.I,.M,.N)) THEN NB_MISS=NB_MISS+1;
                END;
                IF (NB_MISS>1) THEN NB_MISS=1;
        RUN;
        PROC FREQ DATA=BRR_TEMP7 NOPRINT;
                TABLE NB_MISS /OUT=BRR_TEMP8;
                BY &BYVAR;
                WHERE (NB_MISS=0);
        RUN;
        %LET FLAG_STUD=%SCAN(&LIMIT_CRITERIA,1);
        DATA BRR_TEMP8;
                SET BRR_TEMP8;
                IF (COUNT < &FLAG_STUD) THEN FLAG_STUD=1;
                ELSE FLAG_STUD=0;
                KEEP &BYVAR FLAG_STUD;
        RUN;
        PROC SORT DATA=BRR_TEMP7;
                BY &BYVAR &ID_SCHOOL;
        RUN;
        PROC FREQ DATA=BRR_TEMP7 NOPRINT;
                TABLE NB_MISS /OUT=BRR_TEMP9;
                BY &BYVAR &ID_SCHOOL;
                WHERE (NB_MISS=0);
        RUN;
        PROC FREQ DATA=BRR_TEMP9 NOPRINT;
                TABLE NB_MISS /OUT=BRR_TEMP10;
                BY &BYVAR;
        RUN;
```

Box 17.7 [3/3] **SAS® macro of PROC_CORR_NO_PV.sas**

```
        %LET FLAG_SCH=%SCAN(&LIMIT_CRITERIA,2);
        DATA BRR_TEMP10;
                SET BRR_TEMP10;
                IF (COUNT < &FLAG_SCH) THEN FLAG_SCH=1;
                ELSE FLAG_SCH=0;
                KEEP &BYVAR FLAG_SCH;
        RUN;
        PROC SORT DATA=BRR_TEMP7;
                BY &BYVAR NB_MISS;
        RUN;
        PROC FREQ DATA=BRR_TEMP7 NOPRINT;
                TABLE NB_MISS/OUT=BRR_TEMP11;
                BY &BYVAR;
                WEIGHT &REPLI_ROOT.0;
        RUN;
        %LET K=1;
        %LET POPREF=;
        %LET NBVAR=%SCAN(&LIMIT_CRITERIA,4);
        %DO %WHILE(&K <= &NBVAR);
                %LET POPREF_ADD=%SCAN(&BYVAR,&K);
                %LET POPREF=&POPREF &POPREF_ADD;
                %LET K=%EVAL(&K+1);
        %END;

        PROC MEANS DATA=BRR_TEMP11 NOPRINT;
                VAR COUNT;
                BY &POPREF;
                OUTPUT OUT=BRR_TEMP12 SUM=SOMWGT;
        RUN;
        %LET FLAG_PCT=%SCAN(&LIMIT_CRITERIA,3);
        DATA BRR_TEMP13;
                MERGE BRR_TEMP11 BRR_TEMP12;
                BY &POPREF;
                PCT=(COUNT/SOMWGT)*100;
                IF (PCT < &FLAG_PCT) THEN FLAG_PCT=1;
                ELSE FLAG_PCT=0;
                IF (NB_MISS=0);
                KEEP &BYVAR FLAG_PCT;
        RUN;
        DATA &OUTFILE;
                MERGE BRR_TEMP6 BRR_TEMP8 BRR_TEMP10 BRR_TEMP13;
                BY &BYVAR;
        RUN;
        PROC DATASETS LIBRARY=WORK NOLIST;
                DELETE BRR_TEMP7 BRR_TEMP8 BRR_TEMP9 BRR_TEMP10 BRR_TEMP11 BRR_
TEMP12 BRR_TEMP13;
        RUN;

%END;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5 BRR_TEMP6 CORR_
TEMP BRRDATA;
RUN;

OPTIONS NOTES;

%MEND BRR_CORR;
```


PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 17.8 [1/3] **SAS® macro of PROC_CORR_PV.sas**

```
%macro BRR_CORR_PV(INFILE=,
                   REPLI_ROOT=,
                   BYVAR=,
                   EXPLICA=,
                   PV_ROOT=,
                   LIMIT=,
                   LIMIT_CRITERIA=,
                   ID_SCHOOL=,
                   OUTFILE=);

/*

MEANING OF THE MACRO ARGUMENTS

INFILE = INPUT DATA FILE.
REPLI_ROOT = ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL
WEIGHT VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
BYVAR = BREAKDOWN VARIABLES.
EXPLICA = NUMERIC VARIABLE.
PV_ROOT = ROOT OF THE 5 PLAUSIBLE VALUES VARIABLES NAMES
LIMIT = FLAGGING  YES OR NO.
LIMIT_CRITERIA = 1) NUMBER OF STUDENTS 2) NUMBER OF SCHOOLS 3) PERCENTAGE OF
STUDENTS AND 4) NUMBER OF VARIABLES FROM THE BYVAR ARGUMENT FOR DEFINING THE
POPULATION OF REFERENCE.
ID_SCHOOL = VARIABLE NAME FOR THE SCHOOL IDENTIFICATION.
OUTFILE = FILE WITH THE STATISTIC ESTIMATES AND THEIR RESPECTIVE STANDARD ERROR ESTIMATE.

*/

OPTIONS NONOTES;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP1;
RUN;

PROC SORT DATA=&INFILE OUT=BRRDATA(KEEP=&REPLI_ROOT.0-&REPLI_ROOT.80 &BYVAR &PV_
ROOT.1-&PV_ROOT.5 &EXPLICA &ID_SCHOOL);
        BY &BYVAR;
RUN;

%DO I = 0 %TO 80;
        PROC CORR DATA=BRRDATA VARDEF=WGT NOPRINT OUTP=CORR_TEMP;
                VAR &PV_ROOT.1 - &PV_ROOT.5 ;
                WITH &EXPLICA;
        BY &BYVAR;
        WEIGHT &REPLI_ROOT&I;
 RUN;

        DATA CORR_TEMP;
                SET CORR_TEMP;
                L=&I;
        RUN;

        PROC APPEND BASE = BRR_TEMP1 DATA=CORR_TEMP;
        RUN;

%END;

 DATA BRR_TEMP1;
        SET BRR_TEMP1;
        IF (_TYPE_ NE "CORR") THEN DELETE;
        PV1=&PV_ROOT.1;
        PV2=&PV_ROOT.2;
        PV3=&PV_ROOT.3;
        PV4=&PV_ROOT.4;
        PV5=&PV_ROOT.5;
        KEEP &BYVAR L PV1-PV5;
RUN;
```

Box 17.8 [2/3] **SAS® macro of PROC_CORR_PV.sas**

```
DATA BRR_TEMP2(DROP=STAT FIN1-FIN5 MESVAR) BRR_TEMP3(KEEP=&BYVAR STAT FIN1-FIN5
MESVAR);
      SET BRR_TEMP1;
      IF L > 0 THEN OUTPUT BRR_TEMP2;
      ELSE DO;
         STAT =(PV1+PV2+PV3+PV4+PV5)/5;
         FIN1=PV1;
         FIN2=PV2;
         FIN3=PV3;
         FIN4=PV4;
         FIN5=PV5;
         MESVAR=(((STAT-FIN1)**2)+((STAT-FIN2)**2)+((STAT-FIN3)**2)+((STAT-
FIN4)**2)+((STAT-FIN5)**2))/4;
         OUTPUT BRR_TEMP3;
      END;
RUN;
PROC SORT DATA=BRR_TEMP2;
      BY &BYVAR;
RUN;
PROC SORT DATA=BRR_TEMP3;
      BY &BYVAR;
RUN;
DATA BRR_TEMP4;
      MERGE BRR_TEMP2 BRR_TEMP3;
      BY &BYVAR;
      ARRAY A (5)
         PV1-PV5;
      ARRAY B (5)
         FIN1-FIN5;
      ARRAY C(5)
         VAR1-VAR5;
      DO I=1 TO 5;
         C(I)=(1/20)*((A(I)-B(I))**2);
      END;
RUN;
PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
      VAR VAR1 VAR2 VAR3 VAR4 VAR5;
      BY &BYVAR;
      OUTPUT OUT=BRR_TEMP5 SUM=SS1 SS2 SS3 SS4 SS5;
RUN;
DATA BRR_TEMP6;
      MERGE BRR_TEMP3 BRR_TEMP5;
      BY &BYVAR;
      SAMP=(SS1+SS2+SS3+SS4+SS5)/5;
      FINVAR=(SAMP+(1.2*MESVAR));
      SESTAT=(FINVAR)**0.5;
      FORMAT STAT F5.2;
      FORMAT SESTAT F5.2;
      KEEP &BYVAR STAT SESTAT ;
RUN;
%IF (%UPCASE(&LIMIT)=NO) %THEN %DO;

      DATA &OUTFILE;
         SET BRR_TEMP6;
      RUN;
%END;
%ELSE %DO;
      DATA BRR_TEMP7;
         SET BRRDATA;
         NB_MISS=0;
         ARRAY LIST_VAR (6) &EXPLICA &PV_ROOT.1-&PV_ROOT.5;
         DO K=1 TO 6;
                IF (LIST_VAR(K) IN (.,.I,.M,.N)) THEN NB_MISS=NB_MISS+1;
         END;
         IF (NB_MISS>1) THEN NB_MISS=1;
      RUN;
      PROC FREQ DATA=BRR_TEMP7 NOPRINT;
         TABLE NB_MISS /OUT=BRR_TEMP8;
         BY &BYVAR;
         WHERE (NB_MISS=0);
      RUN;
```

Box 17.8 [3/3] **SAS® macro of PROC_CORR_PV.sas**

```
%LET FLAG_STUD=%SCAN(&LIMIT_CRITERIA,1);
        DATA BRR_TEMP8;
                SET BRR_TEMP8;
                IF (COUNT < &FLAG_STUD) THEN FLAG_STUD=1;
                ELSE FLAG_STUD=0;
                KEEP &BYVAR FLAG_STUD;
        RUN;
        PROC SORT DATA=BRR_TEMP7;
                BY &BYVAR &ID_SCHOOL;
        RUN;
        PROC FREQ DATA=BRR_TEMP7 NOPRINT;
                TABLE NB_MISS /OUT=BRR_TEMP9;
                BY &BYVAR &ID_SCHOOL;
                WHERE (NB_MISS=0);
        RUN;
        PROC FREQ DATA=BRR_TEMP9 NOPRINT;
                TABLE NB_MISS /OUT=BRR_TEMP10;
                BY &BYVAR;
        RUN;
        %LET FLAG_SCH=%SCAN(&LIMIT_CRITERIA,2);
        DATA BRR_TEMP10;
                SET BRR_TEMP10;
                IF (COUNT < &FLAG_SCH) THEN FLAG_SCH=1;
                ELSE FLAG_SCH=0;
                KEEP &BYVAR FLAG_SCH;
        RUN;
        PROC SORT DATA=BRR_TEMP7;
                BY &BYVAR NB_MISS;
        RUN;
        PROC FREQ DATA=BRR_TEMP7 NOPRINT;
                TABLE NB_MISS/OUT=BRR_TEMP11;
                BY &BYVAR;
                WEIGHT &REPLI_ROOT.0;
        RUN;
        %LET K=1;
        %LET POPREF=;
        %LET NBVAR=%SCAN(&LIMIT_CRITERIA,4);
        %DO %WHILE(&K <= &NBVAR);
                %LET POPREF_ADD=%SCAN(&BYVAR,&K);
                %LET POPREF=&POPREF &POPREF_ADD;
                %LET K=%EVAL(&K+1);
        %END;

        PROC MEANS DATA=BRR_TEMP11 NOPRINT;
                VAR COUNT;
                BY &POPREF;
                OUTPUT OUT=BRR_TEMP12 SUM=SOMWGT;
        RUN;
        %LET FLAG_PCT=%SCAN(&LIMIT_CRITERIA,3);
        DATA BRR_TEMP13;
                MERGE BRR_TEMP11 BRR_TEMP12;
                BY &POPREF;
                PCT=(COUNT/SOMWGT)*100;
                IF (PCT < &FLAG_PCT) THEN FLAG_PCT=1;
                ELSE FLAG_PCT=0;
                IF (NB_MISS=0);
                KEEP &BYVAR FLAG_PCT;
        RUN;
        DATA &OUTFILE;
                MERGE BRR_TEMP6 BRR_TEMP8 BRR_TEMP10 BRR_TEMP13;
                BY &BYVAR;
        RUN;
        PROC DATASETS LIBRARY=WORK NOLIST;
                DELETE BRR_TEMP7 BRR_TEMP8 BRR_TEMP9 BRR_TEMP10 BRR_TEMP11 BRR_
TEMP12 BRR_TEMP13;
        RUN;
%END;
PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5 BRR_TEMP6 CORR_
TEMP BRRDATA;
RUN;
OPTIONS NOTES;
%MEND BRR_CORR_PV;
```

Box 17.9 [1/3] **SAS® macro of PROC_DIF_NO_PV.sas**

```
%MACRO BRR_PROCMEAN_DIF(INFILE =,
                        REPLI_ROOT =,
                        BYVAR =,
                        VAR =,
                        COMPARE =,
                        CATEGORY =,
                        STAT =,
                        OUTFILE =);

/*

MEANING OF THE MACRO ARGUMENTS

INFILE  = INPUT DATA FILE.
REPLI_ROOT = ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL
WEIGHT VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
BYVAR = BREAKDOWN VARIABLES.
VAR = VARIABLES FOR WHIXH THE STATISTIC IS REQUESTED.
COMPARE = BREAKDOWN VARIABLE NAME FOR WICH CATEGORY CONTRASTS ARE REQUESTED.
CATEGORY = LIST OF THE "COMPARE" VARIABLE CATEGORIES FOR WHICH A CONTRAT IS
REQUESTED.
STAT =  REQUESTED STATISTIQUE.
        SUMWGT = SUM OF THE WEIGHT
        MEAN = MEAN
        VAR = VARIANCE
        STD = STANDARD DEVIATION
        CV = COEFFICIENT OF VARIATION
        SKEWNESS = SKEWNESS
        KURTOSIS = KURTOSIS
        MEDIAN = MEDIAN
        Q1 = FIRST QUARTILE
        Q3 = THIRD QUARTILE
        QRANGE = RANGE BETWEEN Q1 AND Q3
        PX = PERCENTILE, WITH X BETWEEN 1 AND 99
OUTFILE = FILE WITH THE STATISTIC ESTIMATES AND THEIR STANDARD ERROR ESTIMATES.

*/
OPTIONS NONOTES;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP0 BRR_TEMP1 ;
RUN;

PROC SORT DATA=&INFILE OUT=BRRDATA(KEEP=&REPLI_ROOT.0-&REPLI_ROOT.80 &BYVAR
&COMPARE &VAR);
        BY &BYVAR &COMPARE;
RUN;

%DO I = 0 %TO 80;

        PROC MEANS DATA=BRRDATA VARDEF=WGT NOPRINT;
                VAR &VAR ;
                BY &BYVAR &COMPARE;
                WEIGHT &REPLI_ROOT&I;
                OUTPUT OUT=MEAN_TEMP &STAT=PV;
        RUN;

        DATA MEAN_TEMP;
                SET MEAN_TEMP;
                L=&I;
        RUN;

        PROC APPEND BASE = BRR_TEMP0 DATA=MEAN_TEMP;
        RUN;
%END;
```

Box 17.9 [2/3] **SAS® macro of PROC_DIF_NO_PV.sas**

```
%LET DEBUT=1;
%LET SUIVANT=2;
%LET COMPTE=1;

%LET I=1;

%DO %WHILE(%LENGTH(%SCAN(&CATEGORY,&I)));
        %LET I=%EVAL(&I+1);
%END;

%LET NBCAT=%EVAL(&I-1);
%LET NBDIF=%EVAL((&NBCAT*(&NBCAT-1))/2);
%LET ORDRE=1;

%DO J=1 %TO &NBDIF;

        %DO K=&DEBUT %TO &NBCAT-1;

                %LET CAT1=%SCAN(&CATEGORY,&DEBUT);
                %LET CAT2=%SCAN(&CATEGORY,&SUIVANT);

                DATA BRR_DIF1;
                        SET BRR_TEMP0 (RENAME=(PV=M1PV));
                        LENGTH CONTRAST $5;
                        CONTRAST="&CAT1.-&CAT2";
                        IF (&COMPARE=&CAT1);
                        KEEP &BYVAR L CONTRAST M1PV &compare;
                RUN;

                DATA BRR_DIF2 ;
                        SET BRR_TEMP0 (RENAME=(PV=M2PV));
                     IF (&COMPARE=&CAT2);
                        KEEP &BYVAR L M2PV &compare;
                RUN;

                PROC SORT DATA=BRR_DIF1;
                       BY &BYVAR L;
                RUN;

                PROC SORT DATA=BRR_DIF2;
                       BY &BYVAR L;
                RUN;

                DATA BRR_TEMP_DIF;
                       MERGE BRR_DIF1 BRR_DIF2;
                       BY &BYVAR L;
                       IF (M1PV EQ . OR M2PV EQ .) THEN DELETE;
                       PV=M1PV-M2PV;
                       ORDRE=&ORDRE;
                       KEEP &BYVAR CONTRAST L PV ORDRE;
                RUN;

                PROC APPEND BASE = BRR_TEMP1 DATA=BRR_TEMP_DIF;
                RUN;

                %LET SUIVANT=%EVAL(&SUIVANT+1);
                %LET COMPTE=%EVAL(&COMPTE+1);
                %LET ORDRE=%EVAL(&ORDRE+1);

        %END;

        %LET DEBUT=%EVAL(&DEBUT+1);
        %LET SUIVANT=%EVAL(&DEBUT+1);

%END;
```

Box 17.9 [3/3] **SAS® macro of PROC_DIF_NO_PV.sas**

```
DATA BRR_TEMP2(KEEP=&BYVAR CONTRAST PV) BRR_TEMP3(KEEP=&BYVAR CONTRAST ORDRE STAT);
        SET BRR_TEMP1;
        IF L > 0 THEN OUTPUT BRR_TEMP2;
        ELSE DO;
                STAT =PV;
                OUTPUT BRR_TEMP3;
        END;
RUN;

PROC SORT DATA=BRR_TEMP2;
        BY &BYVAR CONTRAST;
RUN;

PROC SORT DATA=BRR_TEMP3;
        BY &BYVAR CONTRAST;
RUN;

DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY &BYVAR CONTRAST;
        VARI=((PV-STAT)**2)*(1/20);
RUN;

PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
        VAR VARI;
        BY &BYVAR CONTRAST;
        OUTPUT OUT=BRR_TEMP5 SUM=SS;
RUN;

DATA BRR_TEMP6;
        MERGE BRR_TEMP3 BRR_TEMP5;
        BY &BYVAR CONTRAST;
        SESTAT=(SS)**0.5;
        FORMAT STAT F10.2;
        FORMAT SESTAT F10.2;
        KEEP &BYVAR CONTRAST ORDRE STAT SESTAT;
RUN;

PROC SORT DATA=BRR_TEMP6 OUT=&OUTFILE (DROP=ORDRE);
        BY &BYVAR ORDRE;
RUN;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP0 BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5 BRR_TEMP6
        MEAN_TEMP BRRDATA BRR_DIF1 BRR_DIF2 BRR_TEMP_DIF;
RUN;

OPTIONS NOTES;

%MEND BRR_PROCMEAN_DIF;
```

Box 17.10 [1/3] **SAS® macro of PROC_DIF_PV.sas**

```
%MACRO BRR_PROCMEAN_DIF_PV(INFILE =,
                           REPLI_ROOT =,
                           BYVAR =,
                           PV_ROOT =,
                           COMPARE =,
                           CATEGORY =,
                           STAT =,
                           OUTFILE =);

/*

MEANING OF THE MACRO ARGUMENTS

INFILE  =       INPUT DATA FILE.
REPLI_ROOT = ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL WEI-
GHT VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
BYVAR = BREAKDOWN VARIABLES.
PV_ROOT =       ROOT OF THE 5 PLAUSIBLE VALUES VARIABLES NAMES.
COMPARE = BREAKDOWN VARIABLE NAME FOR WICH CATEGORY CONTRASTS ARE REQUESTED.
CATEGORY        =       LIST OF THE "COMPARE" VARIABLE CATEGORIES FOR WHICH A
CONTRAT IS REQUESTED.
STAT =  REQUESTED STATISTIQUE.
        SUMWGT = SUM OF THE WEIGHT
        MEAN = MEAN
        VAR = VARIANCE
        STD = STANDARD DEVIATION
        CV =    COEFFICIENT OF VARIATION
        SKEWNESS = SKEWNESS
        KURTOSIS = KURTOSIS
        MEDIAN = MEDIAN
        Q1 =    FIRST QUARTILE
        Q3 =    THIRD QUARTILE
        QRANGE = RANGE BETWEEN Q1 AND Q3
        PX =    PERCENTILE, WITH X BETWEEN 1 AND 99
OUTFILE =       FILE WITH THE STATISTIC ESTIMATES AND THEIR STANDARD ERROR ESTIMATE

*/

OPTIONS NONOTES;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP0 BRR_TEMP1 ;
RUN;

PROC SORT DATA=&INFILE OUT=BRRDATA(KEEP=&REPLI_ROOT.0-&REPLI_ROOT.80 &BYVAR &COM-
PARE &PV_ROOT.1-&PV_ROOT.5);
        BY &BYVAR &COMPARE;
RUN;

%DO I = 0 %TO 80;

        PROC MEANS DATA=BRRDATA VARDEF=WGT NOPRINT;
                VAR &PV_ROOT.1 - &PV_ROOT.5 ;
                BY &BYVAR &COMPARE;
                WEIGHT &REPLI_ROOT&I;
                OUTPUT OUT=MEAN_TEMP &STAT=PV1 - PV5;
        RUN;

        DATA MEAN_TEMP;
                SET MEAN_TEMP;
                L=&I;
        RUN;

        PROC APPEND BASE = BRR_TEMP0 DATA=MEAN_TEMP;
        RUN;

%END;
```


PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 17.10 [2/3] **SAS® macro of PROC_DIF_PV.sas**

```
%LET DEBUT=1;
%LET SUIVANT=2;
%LET COMPTE=1;
%LET I=1;

%DO %WHILE(%LENGTH(%SCAN(&CATEGORY,&I)));
        %LET I=%EVAL(&I+1);
%END;

%LET NBCAT=%EVAL(&I-1);
%LET NBDIF=%EVAL((&NBCAT*(&NBCAT-1))/2);
%LET ORDRE=1;

%DO J=1 %TO &NBDIF;
        %DO K=&DEBUT %TO &NBCAT-1;
                %LET CAT1=%SCAN(&CATEGORY,&DEBUT);
                %LET CAT2=%SCAN(&CATEGORY,&SUIVANT);

                DATA BRR_DIF1;
                        SET BRR_TEMP0 (RENAME=(PV1=M1PV1 PV2=M1PV2 PV3=M1PV3
PV4=M1PV4 PV5=M1PV5));
                        LENGTH CONTRAST $5;
                        CONTRAST="&CAT1.-&CAT2";
                        IF (&COMPARE=&CAT1);
                        KEEP &BYVAR L CONTRAST M1PV1-M1PV5 &compare;
                RUN;

                DATA BRR_DIF2 ;
                        SET BRR_TEMP0 (RENAME=(PV1=M2PV1 PV2=M2PV2 PV3=M2PV3
PV4=M2PV4 PV5=M2PV5));
                     IF (&COMPARE=&CAT2);
                        KEEP &BYVAR L M2PV1-M2PV5 &compare;
                RUN;

                PROC SORT DATA=BRR_DIF1;
                        BY &BYVAR L;
                RUN;

                PROC SORT DATA=BRR_DIF2;
                        BY &BYVAR L;
                RUN;

                DATA BRR_TEMP_DIF;
                        MERGE BRR_DIF1 BRR_DIF2;
                        BY &BYVAR L;
                        IF (M1PV1 EQ . OR M2PV1 EQ .) THEN DELETE;
                        PV1=M1PV1-M2PV1;
                        PV2=M1PV2-M2PV2;
                        PV3=M1PV3-M2PV3;
                        PV4=M1PV4-M2PV4;
                        PV5=M1PV5-M2PV5;
                        ORDRE=&ORDRE;
                        KEEP &BYVAR CONTRAST L PV1-PV5 ORDRE;
                RUN;

                PROC APPEND BASE = BRR_TEMP1 DATA=BRR_TEMP_DIF;
                RUN;

                %LET SUIVANT=%EVAL(&SUIVANT+1);
                %LET COMPTE=%EVAL(&COMPTE+1);
                %LET ORDRE=%EVAL(&ORDRE+1);

        %END;

%LET DEBUT=%EVAL(&DEBUT+1);
%LET SUIVANT=%EVAL(&DEBUT+1);

%END;
```

Box 17.10 [3/3] **SAS® macro of PROC_DIF_PV.sas**

```
DATA BRR_TEMP2(DROP=STAT FIN1-FIN5 MESVAR) BRR_TEMP3(KEEP=&BYVAR CONTRAST ORDRE
STAT FIN1-FIN5 MESVAR);
        SET BRR_TEMP1;
        IF L > 0 THEN OUTPUT BRR_TEMP2;
        ELSE DO;
                STAT =(PV1+PV2+PV3+PV4+PV5)/5;
                FIN1=PV1;
                FIN2=PV2;
                FIN3=PV3;
                FIN4=PV4;
                FIN5=PV5;
                MESVAR=(((STAT-FIN1)**2)+((STAT-FIN2)**2)+((STAT-FIN3)**2)+((STAT-
FIN4)**2)+((STAT-FIN5)**2))/4;
                OUTPUT BRR_TEMP3;
   END;
RUN;

PROC SORT DATA=BRR_TEMP2;
        BY &BYVAR CONTRAST;
RUN;

PROC SORT DATA=BRR_TEMP3;
        BY &BYVAR CONTRAST;
RUN;

DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY &BYVAR CONTRAST;
        ARRAY A (5)
                PV1-PV5;
        ARRAY B (5)
                FIN1-FIN5;
        ARRAY C(5)
                VAR1-VAR5;
        DO I=1 TO 5;
                C(I)=(1/20)*((A(I)-B(I))**2);
        END;
RUN;

PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
        VAR VAR1 VAR2 VAR3 VAR4 VAR5;
        BY &BYVAR CONTRAST;
        OUTPUT OUT=BRR_TEMP5 SUM=SS1 SS2 SS3 SS4 SS5;
RUN;

DATA BRR_TEMP6;
        MERGE BRR_TEMP3 BRR_TEMP5;
        BY &BYVAR CONTRAST;
        SAMP=(SS1+SS2+SS3+SS4+SS5)/5;
        FINVAR=(SAMP+(1.2*MESVAR));
        SESTAT=(FINVAR)**0.5;
        KEEP &BYVAR CONTRAST ORDRE STAT SESTAT;
RUN;

PROC SORT DATA=BRR_TEMP6 OUT=&OUTFILE (DROP=ORDRE);
        BY &BYVAR ORDRE;
RUN;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP0 BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5 BRR_TEMP6
        MEAN_TEMP BRRDATA BRR_DIF1 BRR_DIF2 BRR_TEMP_DIF;
RUN;

OPTIONS NOTES;

%MEND BRR_PROCMEAN_DIF_PV;
```

Box 17.11 [1/6] **SAS® macro of QUARTILE_PV.sas**

```
%MACRO QUARTILE_PV(INFILE=,
                   REPLI_ROOT=,
                   BYVAR = ,
                   PV_ROOT =,
                   INDEX =,
                   LIMIT=,
                   LIMIT_CRITERIA=,
                   ID_SCHOOL=,
                   OUTFILE =);
/*

MEANING OF THE MACRO ARGUMENTS

INFILE= INPUT DATA FILE.
REPLI_ROOT = ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL WEIGHT
VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
BYVAR = BREAKDOWN VARIABLES
PV_ROOT = ROOT OF THE 5 PLAUSIBLE VALUES VARIABLES NAMES
INDEX = VARIABLE NAME USED TO CREATE THE FOUR QUARTERS
LIMIT = FLAGGING  YES OR NO
LIMIT_CRITERIA = 1) NUMBER OF STUDENTS 2) NUMBER OF SCHOOLS 3) PERCENTAGE OF STUDENTS
AND 4) NUMBER OF VARIABLES FROM THE BYVAR ARGUMENT FOR DEFINING THE POPULATION OF
REFERENCE.
ID_SCHOOL =   VARIABLE NAME FOR THE SCHOOL IDENTIFICATION
OUTFILE = FILE WITH THE STATISTIC ESTIMATES AND THEIR STANDARD ERROR ESTIMATES.

*/

OPTIONS NONOTES;

DATA QUARTILE_TEMP1 (KEEP=&BYVAR &REPLI_ROOT.0-&REPLI_ROOT.80 &PV_ROOT.1-&PV_ROOT.5
INDEX1-INDEX5 &ID_SCHOOL &INDEX NB_MISS) ;
      SET &INFILE;
      NB_MISS=0;
      ARRAY A (6) &INDEX &PV_ROOT.1-&PV_ROOT.5;
      DO I=1 TO 6;
          IF (A(I) IN (.,.M,.N,.I)) THEN NB_MISS=NB_MISS+1;
      END;
      INDEX1=&INDEX + (0.01*normal(-01));
      INDEX2=&INDEX + (0.01*normal(-23));
      INDEX3=&INDEX + (0.01*normal(-45));
      INDEX4=&INDEX + (0.01*normal(-67));
      INDEX5=&INDEX + (0.01*normal(-89));
RUN;

PROC SORT  DATA=QUARTILE_TEMP1;
      BY &BYVAR;
RUN;

PROC MEANS DATA=QUARTILE_TEMP1 NOPRINT;
      VAR INDEX1-INDEX5;
      BY &BYVAR;
      WEIGHT &REPLI_ROOT.0;
      WHERE (NB_MISS=0);
      OUTPUT OUT=QUARTILE_TEMP2 P25=Q1_1-Q1_5 P50=Q2_1-Q2_5 P75=Q3_1-Q3_5;
RUN;
```

Box 17.11 [2/6] **SAS® macro of QUARTILE_PV.sas**

```
DATA
      QUARTILE_TEMP3 (KEEP = &BYVAR &REPLI_ROOT.0-&REPLI_ROOT.80 &PV_ROOT.1-&PV_ROOT.5
          INDEX1-INDEX5 CAT1-CAT5 &ID_SCHOOL &INDEX NB_MISS)
      QUARTILE_TEMP4 (KEEP = &BYVAR CAT1-CAT5 &REPLI_ROOT.0 NB_MISS &ID_SCHOOL
&INDEX);
      MERGE QUARTILE_TEMP1 QUARTILE_TEMP2;
      BY &BYVAR;
      ARRAY A1 (5) INDEX1-INDEX5;
      ARRAY A2 (5) Q1_1-Q1_5;
      ARRAY A3 (5) Q2_1-Q2_5;
      ARRAY A4 (5) Q3_1-Q3_5;
      ARRAY A5 (5) CAT1-CAT5;
      DO I=1 TO 5;
          IF (A1(i) <= A2(i)) THEN A5(i)=1;
          IF (A1(i) >  A2(i) AND A1(i) <= A3(i)) THEN A5(i)=2;
          IF (A1(i) >  A3(i) AND A1(i) <= A4(i)) THEN A5(i)=3;
          IF (A1(i) >  A4(i)) THEN A5(i)=4;
          IF (A1(i) IN (.,.M,.N,.I)) THEN A5(i) = .;
      END;
      IF (NB_MISS=0) THEN OUTPUT QUARTILE_TEMP3;
      OUTPUT QUARTILE_TEMP4;
RUN;

%DO QUARTILE=1 %TO 5;

%SUBQUARTILE1(INFILE =QUARTILE_TEMP3,
              REPLI_ROOT =&REPLI_ROOT,
              SUB_BY_VAR = &BYVAR CAT&QUARTILE ,
              VAR =&INDEX,
              OUTFILE =INDEX&QUARTILE);
RUN;

DATA INDEX&QUARTILE;
      SET INDEX&QUARTILE;
      STAT&QUARTILE=STAT;
      SESTAT&QUARTILE=SESTAT;
      CAT=CAT&QUARTILE;
      KEEP &BYVAR CAT STAT&QUARTILE SESTAT&QUARTILE ;
RUN;

%END;

DATA QUARTILE_TEMP5;
      MERGE INDEX1 INDEX2 INDEX3 INDEX4 INDEX5;
      BY &BYVAR CAT;
      STAT=(STAT1+STAT2+STAT3+STAT4+STAT5)/5;
      VAR_IMP=(((STAT1-STAT)**2)+((STAT2-STAT)**2)+((STAT3-STAT)**2)+((STAT4-
STAT)**2)+((STAT5-STAT)**2))/4;
      VAR_SAMP=((SESTAT1**2)+(SESTAT2**2)+(SESTAT3**2)+(SESTAT4**2)+(SESTAT5**2))/5;
      SESTAT=(VAR_SAMP+(1.2*VAR_IMP))**0.5;
      KEEP &BYVAR CAT STAT SESTAT;
RUN;

%DO QUARTILE=1 %TO 5;

%SUBQUARTILE1(INFILE =QUARTILE_TEMP3,
              REPLI_ROOT =&REPLI_ROOT,
              SUB_BY_VAR = &BYVAR CAT&QUARTILE ,
              VAR =&PV_ROOT&QUARTILE,
              OUTFILE =RESULT&QUARTILE);
RUN;

DATA RESULT&QUARTILE;
      SET RESULT&QUARTILE;
      STAT&QUARTILE=STAT;
      SESTAT&QUARTILE=SESTAT;
      CAT=CAT&QUARTILE;
      KEEP &BYVAR CAT STAT&QUARTILE SESTAT&QUARTILE ;
RUN;

%END;
```

Box 17.11 [3/6] **SAS® macro of QUARTILE_PV.sas**

```
DATA QUARTILE_TEMP6;
      MERGE RESULT1 RESULT2 RESULT3 RESULT4 RESULT5;
      BY &BYVAR CAT;
      STAT=(STAT1+STAT2+STAT3+STAT4+STAT5)/5;
      VAR_IMP=(((STAT1-STAT)**2)+((STAT2-STAT)**2)+((STAT3-STAT)**2)+((STAT4-
STAT)**2)+((STAT5-STAT)**2))/4;
      VAR_SAMP=((SESTAT1**2)+(SESTAT2**2)+(SESTAT3**2)+(SESTAT4**2)+(SESTAT5**2))/5;
      SESTAT=(VAR_SAMP+(1.2*VAR_IMP))**0.5;
      KEEP &BYVAR CAT STAT SESTAT;
RUN;

DATA QUARTILE_TEMP7;
      MERGE  QUARTILE_TEMP5 (RENAME=(STAT=INDEX_STAT SESTAT=INDEX_SESTAT))
             QUARTILE_TEMP6 (RENAME=(STAT=PV_STAT SESTAT=PV_SESTAT))  ;
      BY &BYVAR CAT;
RUN;


%IF (%UPCASE(&LIMIT)=NO) %THEN %DO;

      DATA &OUTFILE;
          SET QUARTILE_TEMP7;;
      RUN;

%END;
%ELSE %DO;
      PROC FREQ DATA=QUARTILE_TEMP3 NOPRINT;
          TABLE &BYVAR * CAT1 /OUT=QUARTILE_TEMP8_1;
          TABLE &BYVAR * CAT2 /OUT=QUARTILE_TEMP8_2;
          TABLE &BYVAR * CAT3 /OUT=QUARTILE_TEMP8_3;
          TABLE &BYVAR * CAT4 /OUT=QUARTILE_TEMP8_4;
          TABLE &BYVAR * CAT5 /OUT=QUARTILE_TEMP8_5;
      RUN;
      %LET FLAG_STUD=%SCAN(&LIMIT_CRITERIA,1);
      DATA QUARTILE_TEMP8;
          MERGE
               QUARTILE_TEMP8_1 (RENAME=(COUNT=C1 CAT1=CAT))
               QUARTILE_TEMP8_2 (RENAME=(COUNT=C2 CAT2=CAT))
               QUARTILE_TEMP8_3 (RENAME=(COUNT=C3 CAT3=CAT))
               QUARTILE_TEMP8_4 (RENAME=(COUNT=C4 CAT4=CAT))
               QUARTILE_TEMP8_5 (RENAME=(COUNT=C5 CAT5=CAT));
          BY &BYVAR ;
          C=(C1+C2+C3+C4+C5)/5;
          IF (C < &FLAG_STUD) THEN FLAG_STUD=1;
          ELSE FLAG_STUD=0;
          KEEP &BYVAR FLAG_STUD;
      RUN;

      %DO M=1 %TO 5;
      PROC FREQ DATA=QUARTILE_TEMP3 NOPRINT;
          TABLE &BYVAR * CAT&M * &ID_SCHOOL /OUT=QUARTILE_TEMP9_&M;
      RUN;
      PROC FREQ DATA=QUARTILE_TEMP9_&M NOPRINT;
          TABLE &BYVAR * CAT&M /OUT=QUARTILE_TEMP10_&M;
      RUN;
      %END;
      %LET FLAG_SCH=%SCAN(&LIMIT_CRITERIA,2);
      DATA QUARTILE_TEMP10;
          MERGE
               QUARTILE_TEMP10_1 (RENAME=(COUNT=C1 CAT1=CAT))
               QUARTILE_TEMP10_2 (RENAME=(COUNT=C2 CAT2=CAT))
               QUARTILE_TEMP10_3 (RENAME=(COUNT=C3 CAT3=CAT))
               QUARTILE_TEMP10_4 (RENAME=(COUNT=C4 CAT4=CAT))
               QUARTILE_TEMP10_5 (RENAME=(COUNT=C5 CAT5=CAT));
          BY &BYVAR CAT;
          C=(C1+C2+C3+C4+C5)/5;
          IF (C < &FLAG_SCH) THEN FLAG_SCH=1;
          ELSE FLAG_SCH=0;
          KEEP &BYVAR FLAG_SCH;
      RUN;
```

Box 17.11 [4/6] **SAS® macro of QUARTILE_PV.sas**

```
%LET K=1;
      %LET POPREF=;
      %LET NBVAR=%SCAN(&LIMIT_CRITERIA,4);
      %DO %WHILE(&K <= &NBVAR);
          %LET POPREF_ADD=%SCAN(&BYVAR,&K);
          %LET POPREF=&POPREF &POPREF_ADD;
          %LET K=%EVAL(&K+1);
      %END;
      %LET FLAG_PCT=%SCAN(&LIMIT_CRITERIA,3);
      PROC FREQ DATA=QUARTILE_TEMP4 NOPRINT;
          TABLE &BYVAR * CAT1 * NB_MISS / OUT=QUARTILE_TEMP11_1;
          TABLE &BYVAR * CAT2 * NB_MISS / OUT=QUARTILE_TEMP11_2;
          TABLE &BYVAR * CAT3 * NB_MISS / OUT=QUARTILE_TEMP11_3;
          TABLE &BYVAR * CAT4 * NB_MISS / OUT=QUARTILE_TEMP11_4;
          TABLE &BYVAR * CAT5 * NB_MISS / OUT=QUARTILE_TEMP11_5;
          WEIGHT W_FSTR0;
      RUN;
      DATA QUARTILE_TEMP11;
          MERGE
               QUARTILE_TEMP11_1 (RENAME=(COUNT=C1 CAT1=CAT))
               QUARTILE_TEMP11_2 (RENAME=(COUNT=C2 CAT2=CAT))
               QUARTILE_TEMP11_3 (RENAME=(COUNT=C3 CAT3=CAT))
               QUARTILE_TEMP11_4 (RENAME=(COUNT=C4 CAT4=CAT))
               QUARTILE_TEMP11_5 (RENAME=(COUNT=C5 CAT5=CAT));
          BY &BYVAR CAT NB_MISS;
      RUN;
      PROC MEANS DATA=QUARTILE_TEMP11 NOPRINT;
          VAR C1-C5;
          BY &BYVAR;
          OUTPUT OUT=QUARTILE_TEMP12 SUM=SOMWGT1-SOMWGT5;
      RUN;

      DATA QUARTILE_TEMP13;
          MERGE QUARTILE_TEMP11 QUARTILE_TEMP12;
          BY &BYVAR;
          IF (NB_MISS=0);
          ARRAY A1 (5) C1-C5;
          ARRAY A2 (5) SOMWGT1-SOMWGT5;
          ARRAY A3 (5) PCT1-PCT5;
          DO I=1 TO 5;
               A3(I)=(A1(I)/A2(I))*100;
          END;
          PCT=(PCT1+PCT2+PCT3+PCT4+PCT5)/5;
          IF (PCT < &FLAG_PCT) THEN FLAG_PCT=1;
          ELSE FLAG_PCT=0;
          KEEP &BYVAR FLAG_PCT;
      RUN;
      DATA &OUTFILE;
          MERGE QUARTILE_TEMP7 QUARTILE_TEMP8 QUARTILE_TEMP10 QUARTILE_TEMP13;
          BY &BYVAR;
      RUN;
      PROC DATASETS LIBRARY=WORK NOLIST;
          DELETE
               QUARTILE_TEMP8_1 QUARTILE_TEMP8_2 QUARTILE_TEMP8_3 QUARTILE_TEMP8_4
QUARTILE_TEMP8_5
               QUARTILE_TEMP8
               QUARTILE_TEMP9_1 QUARTILE_TEMP9_2 QUARTILE_TEMP9_3 QUARTILE_TEMP9_4
QUARTILE_TEMP9_5
               QUARTILE_TEMP10_1 QUARTILE_TEMP10_2 QUARTILE_TEMP10_3 QUARTILE_TEMP10_4
QUARTILE_TEMP10_5
               QUARTILE_TEMP10
               QUARTILE_TEMP11_1 QUARTILE_TEMP11_2 QUARTILE_TEMP11_3 QUARTILE_TEMP11_4
QUARTILE_TEMP11_5
               QUARTILE_TEMP11 QUARTILE_TEMP12 QUARTILE_TEMP13;
      RUN;
%END;
```


PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 17.11 [5/6] **SAS® macro of QUARTILE_PV.sas**

```
PROC DATASETS LIBRARY=WORK NOLIST;
      DELETE  QUARTILE_TEMP1 QUARTILE_TEMP2 QUARTILE_TEMP3 QUARTILE_TEMP4 QUARTILE_
TEMP5 QUARTILE_TEMP6 QUARTILE_TEMP7
              INDEX1 INDEX2 INDEX3 INDEX4 INDEX5
              RESULT1 RESULT2 RESULT3 RESULT4 RESULT5;
RUN;

OPTION NOTES;

%MEND ;


%MACRO SUBQUARTILE1(INFILE =,
                    REPLI_ROOT =,
                    SUB_BY_VAR = ,
                    VAR =,
                    OUTFILE =);

PROC DATASETS LIBRARY=WORK NOLIST;
      DELETE BRR_TEMP1;
RUN;

PROC SORT DATA=&INFILE OUT=BRRDATA(KEEP=&REPLI_ROOT.0-&REPLI_ROOT.80 &SUB_BY_VAR
&VAR);
  BY &SUB_BY_VAR;
RUN;

%DO I = 0 %TO 80;
      PROC MEANS DATA=BRRDATA VARDEF=WGT NOPRINT;
          VAR &VAR ;
          BY &SUB_BY_VAR;
          WEIGHT &REPLI_ROOT&I;
          OUTPUT OUT=MEAN_TEMP MEAN=PV;
      RUN;

      DATA MEAN_TEMP;
          SET MEAN_TEMP;
          L=&I;
      RUN;

      PROC APPEND BASE = BRR_TEMP1 DATA=MEAN_TEMP;
      RUN;
%END;

DATA BRR_TEMP2(KEEP=&SUB_BY_VAR PV)BRR_TEMP3(KEEP=&SUB_BY_VAR STAT);
      SET BRR_TEMP1;
      IF L > 0 THEN OUTPUT BRR_TEMP2;
      ELSE DO;
          STAT =PV;
          OUTPUT BRR_TEMP3;
      END;
RUN;

PROC SORT DATA=BRR_TEMP2;
      BY &SUB_BY_VAR;
RUN;

PROC SORT DATA=BRR_TEMP3;
      BY &SUB_BY_VAR;
RUN;

DATA BRR_TEMP4;
      MERGE BRR_TEMP2 BRR_TEMP3;
      BY &SUB_BY_VAR;
      VARI=((PV-STAT)**2)*(1/20);
RUN;
```

Box 17.11 [6/6] **SAS® macro of QUARTILE_PV.sas**

```
PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
      VAR VARI;
      BY &SUB_BY_VAR;
      OUTPUT OUT=BRR_TEMP5 SUM=SS;
RUN;


DATA &OUTFILE;
      MERGE BRR_TEMP3 BRR_TEMP5;
      BY &SUB_BY_VAR;
      SESTAT=(SS)**0.5;
      FORMAT STAT F10.2;
      FORMAT SESTAT F10.2;
      KEEP &SUB_BY_VAR STAT SESTAT;
RUN;

PROC DATASETS LIBRARY=WORK NOLIST;
      DELETE BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5 BRR_TEMP6 MEAN_TEMP
BRRDATA;
RUN;

%MEND SUBQUARTILE1;
```

Box 17.12 [1/3] **SAS® macro of RELATIVE_RISK_NO_PV.sas**

```
%macro BRR_RR(INFILE=,
              REPLI_ROOT=,
              BYVAR=,
              ANTECEDENT=,
              OUTCOME=,
              LIMIT=,
              LIMIT_CRITERIA=,
              ID_SCHOOL=,
              OUTFILE=);

/*

MEANING OF THE MACRO ARGUMENTS

INFILE  =       INPUT DATA FILE.
REPLI_ROOT =    ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL
WEIGHT VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
BYVAR = BREAKDOWN VARIABLES
ANTECEDENT = ANTECEDENT VARIABLES (0,1)
OUTCOME =       CONSEQUENCE VARIABLES (0,1)
LIMIT = FLAGGING  YES OR NO
LIMIT_CRITERIA = 1) NUMBER OF STUDENTS 2) NUMBER OF SCHOOLS 3) PERCENTAGE OF
STUDENTS AND 4) NUMBER OF VARIABLES FROM THE BYVAR ARGUMENT FOR DEFINING THE
POPULATION OF REFERENCE.
ID_SCHOOL       = VARIABLE NAME FOR THE SCHOOL IDENTIFICATION
OUTFILE = FILE WITH THE STATISTIC ESTIMATES AND THEIR STANDARD ERROR ESTIMATES.

*/

OPTIONS NONOTES;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP;
RUN;

PROC SORT DATA=&INFILE
        OUT=BRRDATA(KEEP=&REPLI_ROOT.0-&REPLI_ROOT.80 &BYVAR &ANTECEDENT &OUTCOME
&ID_SCHOOL);
        BY &BYVAR &ANTECEDENT;
RUN;

%DO I = 0 %TO 80;
        PROC FREQ DATA=BRRDATA NOPRINT;
                TABLE &OUTCOME /OUT=FREQ_TEMP ;
                BY &BYVAR &ANTECEDENT ;
                WEIGHT &REPLI_ROOT&I;
                WHERE (&ANTECEDENT IN (0,1) AND &OUTCOME IN (0,1));
        RUN;

        DATA FREQ_TEMP;
                SET FREQ_TEMP;
                L=&I;
                IF (&OUTCOME=1);
                KEEP &BYVAR L &ANTECEDENT PERCENT;
        RUN;

        PROC APPEND BASE = BRR_TEMP DATA=FREQ_TEMP;
        RUN;
%END;

PROC SORT DATA=BRR_TEMP OUT=BRR_TEMP0;
        BY &BYVAR L &ANTECEDENT ;
RUN;

PROC TRANSPOSE DATA=BRR_TEMP0 OUT=BRR_TEMP1;
        VAR PERCENT;
        BY &BYVAR L;
        ID &ANTECEDENT ;
RUN;
```

Box 17.12 [2/3] **SAS® macro of RELATIVE_RISK_NO_PV.sas**

```
DATA BRR_TEMP1;
        SET BRR_TEMP1;
        PV=(_1/_0);
RUN;

DATA BRR_TEMP2(KEEP=&BYVAR PV)BRR_TEMP3(KEEP=&BYVAR STAT);
        SET BRR_TEMP1;
        IF L > 0 THEN OUTPUT BRR_TEMP2;
        ELSE DO;
                STAT =PV;
                OUTPUT BRR_TEMP3;
        END;
RUN;

PROC SORT DATA=BRR_TEMP2;
        BY &BYVAR;
RUN;

PROC SORT DATA=BRR_TEMP3;
        BY &BYVAR;
RUN;

DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY &BYVAR;
        VARI=((PV-STAT)**2)*(1/20);
RUN;

PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
        VAR VARI;
        BY &BYVAR;
        OUTPUT OUT=BRR_TEMP5 SUM=SS;
RUN;

DATA BRR_TEMP6;
        MERGE BRR_TEMP3 BRR_TEMP5;
        BY &BYVAR;
        SESTAT=(SS)**0.5;
        FORMAT STAT F10.2;
        FORMAT SESTAT F10.2;
        KEEP &BYVAR STAT SESTAT;
RUN;

%IF (%UPCASE(&LIMIT)=NO) %THEN %DO;

        DATA &OUTFILE;
                SET BRR_TEMP6;
        RUN;

%END;
%ELSE %DO;
        DATA BRRDATA;
                SET BRRDATA;
                MIS=0;
                IF (&ANTECEDENT NOT IN (0,1) OR &OUTCOME NOT IN (0,1)) THEN MIS=1;
        RUN;
        PROC FREQ DATA=BRRDATA NOPRINT;
                TABLE MIS /OUT=BRR_TEMP7;
                BY &BYVAR;
                WHERE (MIS=0);
        RUN;
        %LET FLAG_STUD=%SCAN(&LIMIT_CRITERIA,1);
        DATA BRR_TEMP7;
                SET BRR_TEMP7;
                IF (COUNT < &FLAG_STUD) THEN FLAG_STUD=1;
                ELSE FLAG_STUD=0;
                KEEP &BYVAR FLAG_STUD;
        RUN;
```

Box 17.12 [3/3] **SAS® macro of RELATIVE_RISK_NO_PV.sas**

```
PROC SORT DATA=BRRDATA;
                BY &BYVAR  &ID_SCHOOL;
        RUN;
        PROC FREQ DATA=BRRDATA NOPRINT;
                TABLE MIS /OUT=BRR_TEMP8;
                BY &BYVAR  &ID_SCHOOL;
                WHERE (MIS=0);
        RUN;
        PROC FREQ DATA=BRR_TEMP8 NOPRINT;
                TABLE MIS /OUT=BRR_TEMP9;
                BY &BYVAR;
        RUN;
        %LET FLAG_SCH=%SCAN(&LIMIT_CRITERIA,2);
        DATA BRR_TEMP9;
                SET BRR_TEMP9;
                IF (COUNT < &FLAG_SCH) THEN FLAG_SCH=1;
                ELSE FLAG_SCH=0;
                KEEP &BYVAR FLAG_SCH ;
        RUN;

        PROC FREQ DATA=BRRDATA NOPRINT;
                TABLE MIS/OUT=BRR_TEMP10;
                BY &BYVAR;
                WEIGHT &REPLI_ROOT.0;
        RUN;
        %LET K=1;
        %LET POPREF=;
        %LET NBVAR=%SCAN(&LIMIT_CRITERIA,4);
        %DO %WHILE(&K <= &NBVAR);
                %LET POPREF_ADD=%SCAN(&BYVAR,&K);
                %LET POPREF=&POPREF &POPREF_ADD;
                %LET K=%EVAL(&K+1);
        %END;

        PROC MEANS DATA=BRR_TEMP10 NOPRINT;
                VAR COUNT;
                BY &POPREF;
                OUTPUT OUT=BRR_TEMP11 SUM=SOMWGT;
        RUN;
        %LET FLAG_PCT=%SCAN(&LIMIT_CRITERIA,3);
        DATA BRR_TEMP12;
                MERGE BRR_TEMP10 BRR_TEMP11;
                BY &POPREF;
                PCT=(COUNT/SOMWGT)*100;
                IF (PCT < &FLAG_PCT) THEN FLAG_PCT=1;
                ELSE FLAG_PCT=0;
                IF (MIS=0);
                KEEP &BYVAR FLAG_PCT;
        RUN;
        DATA &OUTFILE;
                MERGE BRR_TEMP6 BRR_TEMP7 BRR_TEMP9 BRR_TEMP12;
                BY &BYVAR;
                IF (STAT NE .);
        RUN;

        PROC DATASETS LIBRARY=WORK NOLIST;
                DELETE BRR_TEMP7 BRR_TEMP8 BRR_TEMP9 BRR_TEMP10 BRR_TEMP11 BRR_
TEMP12;
        RUN;
%END;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP BRR_TEMP0 BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5
BRR_TEMP6 FREQ_TEMP BRRDATA;
RUN;

OPTIONS NOTES;

%MEND BRR_RR;
```

Box 17.13 [1/5] **SAS® macro of RELATIVE_RISK_PV.sas**

```
%macro BRR_RR_PV(INFILE=,
                 REPLI_ROOT=,
                 BYVAR=,
                 ANTECEDENT_ROOT=,
                 OUTCOME_ROOT=,
                 LIMIT=,
                 LIMIT_CRITERIA=,
                 ID_SCHOOL=,
                 OUTFILE=);

/*

MEANING OF THE MACRO ARGUMENTS

INFILE  = INPUT DATA FILE.
REPLI_ROOT = ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES.
FINAL WEIGHT VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
BYVAR = BREAKDOWN VARIABLES
ANTECEDENT_ROOT =      ROOT OF THE 5 ANTECEDENT VARIABLES (0,1)
OUTCOME_ROOT = ROOT OF THE 5 CONSEQUENCE VARIABLES   (0,1)
LIMIT = FLAGGING  YES OR NO
LIMIT_CRITERIA = 1) NUMBER OF STUDENTS 2) NUMBER OF SCHOOLS 3) PERCENTAGE OF
STUDENTS AND 4) NUMBER OF VARIABLES FROM THE BYVAR ARGUMENT FOR DEFINING THE
POPULATION OF REFERENCE.
ID_SCHOOL =    VARIABLE NAME FOR THE SCHOOL IDENTIFICATION
OUTFILE = FILE WITH THE STATISTIC ESTIMATES AND THEIR STANDARD ERROR ESTIMATES.

*/

OPTIONS NONOTES;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP;
RUN;

%DO I = 1 %TO 5;

PROC SORT DATA=&INFILE
        OUT=BRRDATA(KEEP=      &REPLI_ROOT.0-&REPLI_ROOT.80 &BYVAR &ANTECEDENT_
ROOT.1-&ANTECEDENT_ROOT.5
                                                &OUTCOME_ROOT.1-&OUTCOME_ROOT.5 &ID_
SCHOOL);

        BY &BYVAR &ANTECEDENT_ROOT&I;
RUN;


%DO J = 0 %TO 80;
        PROC FREQ DATA=BRRDATA NOPRINT;
                TABLE &OUTCOME_ROOT&I /OUT=FREQ_TEMP ;
                BY &BYVAR &ANTECEDENT_ROOT&I ;
                WEIGHT &REPLI_ROOT&J;
                WHERE (&ANTECEDENT_ROOT&I IN (0,1) AND &OUTCOME_ROOT&I IN (0,1));
        RUN;

        DATA FREQ_TEMP;
                SET FREQ_TEMP;
                K=&I;
                L=&J;
                &ANTECEDENT_ROOT=&ANTECEDENT_ROOT&I;
                IF (&OUTCOME_ROOT&I=1);
                KEEP &BYVAR K L &ANTECEDENT_ROOT PERCENT;
        RUN;

        PROC APPEND BASE = BRR_TEMP DATA=FREQ_TEMP;
        RUN;
%END;
```


PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 17.13 [2/5] **SAS® macro of RELATIVE_RISK_PV.sas**

```
PROC SORT DATA=BRR_TEMP OUT=BRR_PREP1;
        BY &BYVAR L K &ANTECEDENT_ROOT ;
RUN;
PROC TRANSPOSE DATA=BRR_PREP1 OUT=BRR_PREP2;
        VAR PERCENT;
        BY &BYVAR L K;
        ID &ANTECEDENT_ROOT ;
RUN;
DATA BRR_PREP2;
        SET BRR_PREP2;
        PV=(_1/_0);
        KEEP &BYVAR L K PV;
RUN;
PROC TRANSPOSE DATA=BRR_PREP2 OUT=BRR_PREP3 PREFIX=PV;
        VAR PV;
        BY &BYVAR L;
        ID K ;
RUN;
DATA BRR_TEMP1;
        SET BRR_PREP3;
        DROP _NAME_ ;
RUN;

DATA BRR_TEMP2(DROP=STAT FIN1-FIN5 MESVAR)BRR_TEMP3(KEEP=&BYVAR STAT FIN1-FIN5
MESVAR);
        SET BRR_TEMP1;
        IF L > 0 THEN OUTPUT BRR_TEMP2;
        ELSE DO;
                STAT =(PV1+PV2+PV3+PV4+PV5)/5;
                FIN1=PV1;
                FIN2=PV2;
                FIN3=PV3;
                FIN4=PV4;
                FIN5=PV5;
                MESVAR=(((STAT-FIN1)**2)+((STAT-FIN2)**2)+((STAT-FIN3)**2)+((STAT-
FIN4)**2)+((STAT-FIN5)**2))/4;
                OUTPUT BRR_TEMP3;
        END;
RUN;

PROC SORT DATA=BRR_TEMP2;
        BY &BYVAR;
RUN;

PROC SORT DATA=BRR_TEMP3;
        BY &BYVAR ;
RUN;

DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY &BYVAR ;
        ARRAY A (5)
                PV1-PV5;
        ARRAY B (5)
                FIN1-FIN5;
        ARRAY C(5)
                VAR1-VAR5;
        DO I=1 TO 5;
                C(I)=(1/20)*((A(I)-B(I))**2);
        END;
RUN;

PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
        VAR VAR1 VAR2 VAR3 VAR4 VAR5;
        BY &BYVAR ;
        OUTPUT OUT=BRR_TEMP5 SUM=SS1 SS2 SS3 SS4 SS5;
RUN;
```

Box 17.13 [3/5] **SAS® macro of RELATIVE_RISK_PV.sas**

```
DATA BRR_TEMP6;
        MERGE BRR_TEMP3 BRR_TEMP5;
        BY &BYVAR ;
        SAMP=(SS1+SS2+SS3+SS4+SS5)/5;
        FINVAR=(SAMP+(1.2*MESVAR));
        SESTAT=(FINVAR)**0.5;
        FORMAT STAT F10.2;
        FORMAT SESTAT F10.2;
        KEEP &BYVAR STAT SESTAT;
RUN;

%IF (%UPCASE(&LIMIT)=NO) %THEN %DO;

        DATA &OUTFILE;
                SET BRR_TEMP6;
        RUN;

%END;
%ELSE %DO;
        DATA BRRDATA;
                SET BRRDATA;
                MIS=1;
                IF (&ANTECEDENT_ROOT.1 IN (0,1) AND &OUTCOME_ROOT.1 IN (0,1)) THEN
MIS=0;
        RUN;
        %DO M=1 %TO 5;
        PROC FREQ DATA=BRRDATA NOPRINT;
                TABLE MIS/OUT=BRR_TEMP7;
                BY &BYVAR;
                WHERE (MIS=0);
        RUN;
        DATA BRR_TEMP8;
                SET BRR_TEMP7;
                COUNT&M=COUNT;
                KEEP &BYVAR COUNT&M;
        RUN;
        %IF (&M=1) %THEN %DO;
                DATA BRR_TEMP9;
                        SET BRR_TEMP8;
                RUN;
        %END;
        %ELSE %DO;
                DATA BRR_TEMP9;
                        MERGE BRR_TEMP9 BRR_TEMP8;
                        BY &BYVAR ;
                RUN;
        %END;
        %END;
        %LET FLAG_STUD=%SCAN(&LIMIT_CRITERIA,1);
        DATA BRR_TEMP9;
                SET BRR_TEMP9;
                COUNT=(COUNT1+COUNT2+COUNT3+COUNT4+COUNT5)/5;
                IF (COUNT < &FLAG_STUD) THEN FLAG_STUD=1;
                ELSE FLAG_STUD=0;
                KEEP &BYVAR FLAG_STUD;
        RUN;

        PROC SORT DATA=BRRDATA;
                BY &BYVAR  &ID_SCHOOL;
        RUN;

        %DO M=1 %TO 5;
        PROC FREQ DATA=BRRDATA NOPRINT;
                TABLE MIS /OUT=BRR_TEMP10;
                BY &BYVAR  &ID_SCHOOL;
                WHERE (MIS=0);
        RUN;
        PROC FREQ DATA=BRR_TEMP10 NOPRINT;
                TABLE MIS  /OUT=BRR_TEMP11;
                BY &BYVAR;
        RUN;
```


PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 17.13 [4/5] **SAS® macro of RELATIVE_RISK_PV.sas**

```
DATA BRR_TEMP12;
                SET BRR_TEMP11;
                COUNT&M=COUNT;
                KEEP &BYVAR COUNT&M;
        RUN;
        %IF (&M=1) %THEN %DO;
                DATA BRR_TEMP13;
                        SET BRR_TEMP12;
                RUN;
        %END;
        %ELSE %DO;
                DATA BRR_TEMP13;
                        MERGE BRR_TEMP13 BRR_TEMP12;
                        BY &BYVAR;
                RUN;
        %END;
        %END;
        %LET FLAG_SCH=%SCAN(&LIMIT_CRITERIA,2);
        DATA BRR_TEMP13;
                SET BRR_TEMP13;
                COUNT=(COUNT1+COUNT2+COUNT3+COUNT4+COUNT5)/5;
                IF (COUNT < &FLAG_SCH) THEN FLAG_SCH=1;
                ELSE FLAG_SCH=0;
                KEEP &BYVAR FLAG_SCH;
        RUN;

        %DO M=1 %TO 5;
        PROC FREQ DATA=BRRDATA NOPRINT;
                TABLE MIS/OUT=BRR_TEMP14;
                BY &BYVAR;
                WEIGHT &REPLI_ROOT.0;
        RUN;
        DATA BRR_TEMP15;
                SET BRR_TEMP14;
                COUNT&M=COUNT;
                KEEP &BYVAR MIS COUNT&M;
        RUN;
        %IF (&M=1) %THEN %DO;
                DATA BRR_TEMP16;
                        SET BRR_TEMP15;
                RUN;
        %END;
        %ELSE %DO;
                DATA BRR_TEMP16;
                        MERGE BRR_TEMP16 BRR_TEMP15;
                        BY &BYVAR MIS;
                RUN;
        %END;
        %END;

        %LET K=1;
        %LET POPREF=;
        %LET NBVAR=%SCAN(&LIMIT_CRITERIA,4);
        %DO %WHILE(&K <= &NBVAR);
                %LET POPREF_ADD=%SCAN(&BYVAR,&K);
                %LET POPREF=&POPREF &POPREF_ADD;
                %LET K=%EVAL(&K+1);
        %END;

        PROC MEANS DATA=BRR_TEMP16 NOPRINT;
                VAR COUNT1-COUNT5 ;
                BY &POPREF;
                OUTPUT OUT=BRR_TEMP17 SUM=SOMWGT1-SOMWGT5;
        RUN;
```

Box 17.13 [5/5] **SAS® macro of RELATIVE_RISK_PV.sas**

```
%LET FLAG_PCT=%SCAN(&LIMIT_CRITERIA,3);
        DATA BRR_TEMP18;
                MERGE BRR_TEMP16 BRR_TEMP17;
                BY &POPREF;
                IF (MIS=0);
                ARRAY B1 (5) PCT1-PCT5;
                ARRAY B2 (5) COUNT1-COUNT5;
                ARRAY B3 (5) SOMWGT1-SOMWGT5;
                DO A=1 TO 5;
                        B1(A)=(B2(A)/B3(A))*100;
                END;
                PCT=(PCT1+PCT2+PCT3+PCT4+PCT5)/5;
                IF (PCT < &FLAG_PCT) THEN FLAG_PCT=1;
                ELSE FLAG_PCT=0;
                KEEP &BYVAR FLAG_PCT;
        RUN;

        DATA &OUTFILE;
                MERGE BRR_TEMP6 BRR_TEMP9 BRR_TEMP13 BRR_TEMP18;
                BY &BYVAR ;
        RUN;

        PROC DATASETS LIBRARY=WORK NOLIST;
                DELETE BRR_TEMP7 BRR_TEMP8 BRR_TEMP9 BRR_TEMP10 BRR_TEMP11 BRR_
TEMP12 BRR_TEMP13
                                BRR_TEMP14 BRR_TEMP15 BRR_TEMP16 BRR_TEMP17 BRR_
TEMP18;
        RUN;

%END;


PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_PREP1 BRR_PREP2 BRR_PREP3 BRR_TEMP BRR_TEMP1 BRR_TEMP2 BRR_TEMP3
BRR_TEMP4 BRR_TEMP5 FREQ_TEMP BRR_TEMP6 MEAN_TEMP BRRDATA;
RUN;

OPTIONS NOTES;

%MEND BRR_RR_PV;
```

Box 17.14 [1/2] **SAS® macro of EFFECT_SIZE_NO_PV.sas**

```
%MACRO BRR_EFFECT(INFILE=,
                  REPLI_ROOT=,
                  BYVAR=,
                  VAR=,
                  EFFECT=,
                  OUTFILE=);


/*

MEANING OF THE MACRO ARGUMENTS

INFILE = INPUT DATA FILE.
REPLI_ROOT = ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL
WEIGHT VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
BYVAR   = BREAKDOWN VARIABLES.
VAR = DEPENDENT VARIABLES.
EFFECT = THIS ARGUMENT HAS THREE COMPONENTS: THE VARIABLE NAME AND THE TWO
CATEGORIES ON WHICH THE EFFECT SIZE WILL BE COMPUTED. THE EFFECT SIZE WILL BE THE
SECOND COMPONENT MINUS THE THIRD COMPONENT.
OUTFILE = FILE WITH THE STATISTIC ESTIMATES AND THEIR STANDARD ERROR ESTIMATES.

*/

%LET GROUP=%SCAN(&EFFECT,1);
%LET CAT1=%SCAN(&EFFECT,2);
%LET CAT2=%SCAN(&EFFECT,3);

OPTIONS NONOTES;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP1;
RUN;

DATA BRRDATA;
        SET &INFILE;
        IF (&GROUP=&CAT1) THEN VAR_GROUP=1;
        IF (&GROUP=&CAT2) THEN VAR_GROUP=2;
        IF (VAR_GROUP IN (1,2));
        KEEP &REPLI_ROOT.0-&REPLI_ROOT.80 &BYVAR &VAR VAR_GROUP;
RUN;

PROC SORT DATA=BRRDATA;
  BY &BYVAR VAR_GROUP;
RUN;

%DO I = 0 %TO 80;
        PROC MEANS DATA=BRRDATA VARDEF=WGT NOPRINT;
                VAR &VAR ;
                BY &BYVAR VAR_GROUP;
                WEIGHT &REPLI_ROOT&I;
                OUTPUT OUT=MEAN_TEMP MEAN=STAT1 VAR=STAT2;
        RUN;

        DATA MEAN_TEMP;
                SET MEAN_TEMP;
                L=&I;
        RUN;

        PROC APPEND BASE = BRR_TEMP1 DATA=MEAN_TEMP;
        RUN;

%END;

PROC SORT DATA=BRR_TEMP1;
        BY &BYVAR L VAR_GROUP;
RUN;
```

Box 17.14 [2/2] **SAS® macro of EFFECT_SIZE_NO_PV.sas**

```
PROC TRANSPOSE DATA=BRR_TEMP1 OUT=BRR_TEMP2 PREFIX=M;
        VAR STAT1       ;
        BY &BYVAR L;
        ID VAR_GROUP;
RUN;
PROC TRANSPOSE DATA=BRR_TEMP1 OUT=BRR_TEMP3 PREFIX=V;
        VAR STAT2       ;
        BY &BYVAR L;
        ID VAR_GROUP;
RUN;
DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY      &BYVAR L;
        PV=(M1-M2)/(((V1+V2)/2)**0.5);
        KEEP  &BYVAR L PV;
RUN;

DATA BRR_TEMP5(KEEP=&BYVAR PV)BRR_TEMP6(KEEP=&BYVAR STAT);
        SET BRR_TEMP4;
        IF L > 0 THEN OUTPUT BRR_TEMP5;
        ELSE DO;
                STAT =PV;
                OUTPUT BRR_TEMP6;
        END;
RUN;

PROC SORT DATA=BRR_TEMP5;
        BY &BYVAR;
RUN;

PROC SORT DATA=BRR_TEMP6;
        BY &BYVAR;
RUN;

DATA BRR_TEMP7;
        MERGE BRR_TEMP5 BRR_TEMP6;
        BY &BYVAR;
        VARI=((PV-STAT)**2)*(1/20);
RUN;

PROC UNIVARIATE DATA=BRR_TEMP7 NOPRINT;
        VAR VARI;
        BY &BYVAR;
        OUTPUT OUT=BRR_TEMP8 SUM=SS;
RUN;

DATA &OUTFILE;
        MERGE BRR_TEMP6 BRR_TEMP8;
        BY &BYVAR;
        SESTAT=(SS)**0.5;
        FORMAT STAT F10.2;
        FORMAT SESTAT F10.2;
        KEEP &BYVAR STAT SESTAT;
RUN;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRRDATA MEAN_TEMP BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5
BRR_TEMP6 BRR_TEMP7 BRR_TEMP8;
RUN;

OPTIONS NOTES;

%MEND;
```

Box 17.15 [1/3] **SAS® macro of EFFECT_SIZE_PV.sas**

```
%MACRO BRR_EFFECT_PV(INFILE=,
                     REPLI_ROOT=,
                     BYVAR=,
                     PV_ROOT=,
                     EFFECT=,
                     OUTFILE=);

/*

MEANING OF THE MACRO ARGUMENTS

INFILE = INPUT DATA FILE.
REPLI_ROOT =  ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL
WEIGHT VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
BYVAR = BREAKDOWN VARIABLES.
PV_ROOT =     ROOT OF THE 5 PLAUSIBLE VALUES VARIABLES NAMES.
EFFECT = THIS ARGUMENT HAS THREE COMPONENTS: THE VARIABLE NAME AND THE TWO
CATEGORIES ON WHICH THE EFFECT        SIZE WILL BE COMPUTED. THE EFFECT SIZE WILL
BE THE SECOND COMPONENT MINUS THE THIRD COMPONENT.
OUTFILE = FILE WITH THE STATISTIC ESTIMATES AND THEIR STANDARD ERROR ESTIMATES.

*/

%LET GROUP=%SCAN(&EFFECT,1);
%LET CAT1=%SCAN(&EFFECT,2);
%LET CAT2=%SCAN(&EFFECT,3);

OPTIONS NONOTES;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP1;
RUN;

DATA BRRDATA;
        SET &INFILE;
        IF (&GROUP=&CAT1) THEN VAR_GROUP=1;
        IF (&GROUP=&CAT2) THEN VAR_GROUP=2;
        IF (VAR_GROUP IN (1,2));
        KEEP &REPLI_ROOT.0-&REPLI_ROOT.80 &PV_ROOT.1-&PV_ROOT.5 &BYVAR  VAR_GROUP;
RUN;

PROC SORT DATA=BRRDATA;
  BY &BYVAR VAR_GROUP;
RUN;

%DO I = 0 %TO 80;
        PROC MEANS DATA=BRRDATA VARDEF=WGT NOPRINT;
                VAR &PV_ROOT.1 - &PV_ROOT.5  ;
                BY &BYVAR VAR_GROUP;
                WEIGHT &REPLI_ROOT&I;
                OUTPUT OUT=MEAN_TEMP MEAN=MOY1-MOY5 VAR=VARI1-VARI5;
        RUN;

        DATA MEAN_TEMP;
                SET MEAN_TEMP;
                L=&I;
        RUN;

        PROC APPEND BASE = BRR_TEMP1 DATA=MEAN_TEMP;
        RUN;

%END;

PROC SORT DATA=BRR_TEMP1;
        BY &BYVAR L VAR_GROUP;
RUN;
```

Box 17.15 [2/3] **SAS® macro of EFFECT_SIZE_PV.sas**

```
PROC TRANSPOSE DATA=BRR_TEMP1 (RENAME=(MOY1=V1 MOY2=V2 MOY3=V3 MOY4=V4 MOY5=V5))
OUT=BRR_TEMP2 PREFIX=M;
        VAR V1-V5       ;
        BY &BYVAR L;
        ID VAR_GROUP;
RUN;
PROC TRANSPOSE DATA=BRR_TEMP1 (RENAME=(VARI1=V1 VARI2=V2 VARI3=V3 VARI4=V4
VARI5=V5))OUT=BRR_TEMP3 PREFIX=V;
        VAR V1-V5       ;
        BY &BYVAR L;
        ID VAR_GROUP;
RUN;
DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY      &BYVAR L _NAME_;
        PV=(M1-M2)/(((V1+V2)/2)**0.5);
        KEEP  &BYVAR L PV _NAME_;
RUN;
PROC TRANSPOSE DATA=BRR_TEMP4 OUT=BRR_TEMP5(DROP=_NAME_) PREFIX=P;
        VAR PV;
        BY &BYVAR L;
        ID _NAME_;
RUN;

DATA BRR_TEMP6(DROP=STAT FIN1-FIN5 MESVAR) BRR_TEMP7(KEEP=&BYVAR STAT FIN1-FIN5
MESVAR);
        SET BRR_TEMP5;
    IF L > 0 THEN OUTPUT BRR_TEMP6;
    ELSE DO;
        STAT =(PV1+PV2+PV3+PV4+PV5)/5;
                FIN1=PV1;
                FIN2=PV2;
                FIN3=PV3;
                FIN4=PV4;
                FIN5=PV5;
        MESVAR=(((STAT-FIN1)**2)+((STAT-FIN2)**2)+((STAT-FIN3)**2)+((STAT-
FIN4)**2)+((STAT-FIN5)**2))/4;
        OUTPUT BRR_TEMP7;
        END;
RUN;

PROC SORT DATA=BRR_TEMP6;
        BY &BYVAR;
RUN;

PROC SORT DATA=BRR_TEMP7;
        BY &BYVAR;
RUN;

DATA BRR_TEMP8;
        MERGE BRR_TEMP6 BRR_TEMP7;
        BY &BYVAR;
        ARRAY A (5)
                PV1-PV5;
        ARRAY B (5)
                FIN1-FIN5;
        ARRAY C(5)
                VAR1-VAR5;
        DO I=1 TO 5;
        C(I)=(1/20)*((A(I)-B(I))**2);
        END;
RUN;

PROC UNIVARIATE DATA=BRR_TEMP8 NOPRINT;
        VAR VAR1 VAR2 VAR3 VAR4 VAR5;
        BY &BYVAR;
        OUTPUT OUT=BRR_TEMP9 SUM=SS1 SS2 SS3 SS4 SS5;
RUN;
```

Box 17.15 [3/3] **SAS® macro of EFFECT_SIZE_PV.sas**

```
DATA &OUTFILE;
        MERGE BRR_TEMP7 BRR_TEMP9;
        BY &BYVAR;
        SAMP=(SS1+SS2+SS3+SS4+SS5)/5;
        FINVAR=(SAMP+(1.2*MESVAR));
        SESTAT=(FINVAR)**0.5;
        FORMAT STAT F10.2;
        FORMAT SESTAT F10.2;
        KEEP &BYVAR STAT SESTAT  ;
RUN;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRRDATA MEAN_TEMP BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5
BRR_TEMP6 BRR_TEMP7 BRR_TEMP8 BRR_TEMP9;
RUN;

OPTIONS NOTES;

%MEND;
```

Box 17.16 [1/5] **SAS® macro of PROC_MIXED_NO_PV.sas**

```
%macro BRR_MIXED(INFILE=,
                 REPLI_ROOT=,
                 VARDEP=,
                 FIXEF=,
                 RANEF=,
                 BYVAR=,
                 LEVEL2=,
                 OUTSCREEN=,
                 OUTFILE=);

/*

MEANING OF THE MACRO ARGUMENTS

INFILE  =       INPUT DATA FILE.
REPLI_ROOT =    ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL
WEIGHT VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
VARDEP =        DEPENDENT VARIABLES.
FIXEF = LIST OF INDEPENDENT VARIABLES WITH FIXED EFFECT.
RANEF = LIST OF INDEPENDENT VARIABLES WITH RANDOM EFFECT.
LEVEL2  =       LEVEL 2 IDENTIFICATION VARIABLE.
BYVAR = BREAKDOWN VARIABLES.
OUTSCREEN       =       DOS ADRESS FOR EXPORTING LISTING.
OUTFILE =       FILE WITH THE STATISTIC ESTIMATES AND THEIR STANDARD ERROR
ESTIMATES.

*/


OPTION NONOTES CLEANUP;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP0;
RUN;

PROC SORT DATA=&INFILE
        OUT=BRR_PREP1(KEEP=&REPLI_ROOT.0-&REPLI_ROOT.80  &VARDEP &BYVAR &FIXEF
&RANEF &LEVEL2);
        BY &BYVAR;
RUN;

%LET I=1;
%DO %WHILE(%LENGTH(%SCAN(&FIXEF,&I)));
        %LET I=%EVAL(&I+1);
%END;
%LET NB_FIXE=%EVAL(&I-1);


%LET I=1;
%DO %WHILE(%LENGTH(%SCAN(&RANEF,&I)));
        %LET I=%EVAL(&I+1);
%END;
%LET NB_RAN=%EVAL(&I-1);
%LET NB_TOT=%EVAL(&NB_FIXE+&NB_RAN+1);

DATA BRR_PREP2;
                SET BRR_PREP1;
                NB_MISS=0;
                ARRAY LIST_VAR (&NB_TOT) &VARDEP &FIXEF &RANEF;
                DO K=1 TO &NB_TOT;
                        IF (LIST_VAR(K) IN (.,.I,.M,.N)) THEN NB_MISS=NB_MISS+1;
                END;
                IF (NB_MISS>1) THEN NB_MISS=1;
RUN;

PROC FREQ DATA=BRR_PREP2 NOPRINT;
        TABLE NB_MISS/OUT=&OUTFILE._DELETION;
        BY &BYVAR;
        WEIGHT &REPLI_ROOT.0;
RUN;
```

Box 17.16 [2/5] **SAS® macro of PROC_MIXED_NO_PV.sas**

```
PROC UNIVARIATE DATA=BRR_PREP2 NOPRINT;
        VAR &REPLI_ROOT.0;
        BY &BYVAR;
        WHERE (NB_MISS=0);
        OUTPUT OUT=BRR_PREP3 N=NBRE SUM=SOMWGT;
RUN;

DATA BRRDATA;
        MERGE BRR_PREP2 BRR_PREP3;
        BY &BYVAR;
        ARRAY OLDWGT (81)&REPLI_ROOT.0-&REPLI_ROOT.80;
    ARRAY NEWWGT (81) STD_WGT0-STD_WGT80;
        DO I=1 TO 81;
                NEWWGT(I)=(OLDWGT(I)/SOMWGT)*NBRE;
        END;
RUN;

%DO J = 0 %TO 80;

                PROC MIXED DATA=BRRDATA NOCLPRINT NOITPRINT NOINFO METHOD=ML;
                        CLASS &LEVEL2;
                        MODEL &VARDEP=&FIXEF &RANEF /SOLUTION;
                        RANDOM INTERCEPT &RANEF /SUBJECT=&LEVEL2;
                        WEIGHT STD_WGT&J;
                        BY &BYVAR;
                        ODS OUTPUT SOLUTIONF=FIXE_TEMP COVPARMS=COV_TEMP;
                RUN;

                DATA FIXE_TEMP;
                        SET FIXE_TEMP;
                        L=&J;
                        KEEP &BYVAR L EFFECT ESTIMATE ;
                RUN;
                PROC SORT DATA=FIXE_TEMP;
                        by &BYVAR L EFFECT;
                RUN;
                DATA COV_TEMP;
                        SET COV_TEMP;
                        L=&J;
                        KEEP &BYVAR L COVPARM ESTIMATE ;
                RUN;
                PROC SORT DATA=COV_TEMP;
                        BY &BYVAR L COVPARM;
                RUN;

                %IF (&J=0) %then %do;
                        DATA BRR_TEMP_FIXE;
                                SET FIXE_TEMP;
                        RUN;
                        DATA BRR_TEMP_VAR;
                                SET COV_TEMP;
                        RUN;
                %END;

                DATA BRR_TEMP_FIXE;
                        MERGE BRR_TEMP_FIXE FIXE_TEMP;
                        BY &BYVAR L EFFECT;
                RUN;
                DATA BRR_TEMP_VAR;
                        MERGE BRR_TEMP_VAR COV_TEMP;
                        BY &BYVAR L COVPARM;
                RUN;

%END;

/* FIXE PARAMETERS */

PROC SORT DATA=BRR_TEMP_FIXE OUT=BRR_TEMP1;
        BY &BYVAR L EFFECT;
RUN;
```

Box 17.16 [3/5] **SAS® macro of PROC_MIXED_NO_PV.sas**

```
DATA BRR_TEMP2(KEEP=&BYVAR EFFECT ESTIMATE) BRR_TEMP3(KEEP=&BYVAR EFFECT STAT);
        SET BRR_TEMP1;
        IF L > 0 THEN OUTPUT BRR_TEMP2;
        ELSE DO;
                STAT =ESTIMATE;
                OUTPUT BRR_TEMP3;
        END;
RUN;

PROC SORT DATA=BRR_TEMP2;
        BY &BYVAR EFFECT;
RUN;

PROC SORT DATA=BRR_TEMP3;
        BY &BYVAR EFFECT;
RUN;

DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY &BYVAR EFFECT;
        VARI=((ESTIMATE-STAT)**2)*(1/20);
RUN;

PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
        VAR VARI;
        BY &BYVAR EFFECT;
        OUTPUT OUT=BRR_TEMP5 SUM=SS;
RUN;

DATA &OUTFILE._FIXE;
        MERGE BRR_TEMP3 BRR_TEMP5;
        BY &BYVAR EFFECT;
        SESTAT=(SS)**0.5;
        FORMAT STAT F10.2;
        FORMAT SESTAT F10.2;
        KEEP &BYVAR EFFECT STAT SESTAT;
RUN;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_PREP1 BRR_PREP2 BRR_PREP3 BRR_TEMP_FIXE FIXE_TEMP BRR_TEMP1
BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5 MEAN_TEMP BRRDATA;
RUN;

/* RANDOM PARAMETERS */

PROC SORT DATA=BRR_TEMP_VAR OUT=BRR_TEMP1;
        BY &BYVAR L COVPARM;
RUN;

DATA BRR_TEMP2(KEEP=&BYVAR COVPARM ESTIMATE) BRR_TEMP3(KEEP=&BYVAR COVPARM STAT);
        SET BRR_TEMP1;
        IF L > 0 THEN OUTPUT BRR_TEMP2;
        ELSE DO;
                STAT =ESTIMATE;
                OUTPUT BRR_TEMP3;
        END;
RUN;

PROC SORT DATA=BRR_TEMP2;
        BY &BYVAR COVPARM;
RUN;

PROC SORT DATA=BRR_TEMP3;
        BY &BYVAR COVPARM;
RUN;

DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY &BYVAR COVPARM;
        VARI=((ESTIMATE-STAT)**2)*(1/20);
RUN;
```


PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 17.16 [4/5] **SAS® macro of PROC_MIXED_NO_PV.sas**

```
PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
        VAR VARI;
        BY &BYVAR COVPARM;
        OUTPUT OUT=BRR_TEMP5 SUM=SS;
RUN;

DATA &OUTFILE._VARIANCE;
        MERGE BRR_TEMP3 BRR_TEMP5;
        BY &BYVAR COVPARM;
        SESTAT=(SS)**0.5;
        FORMAT STAT F10.2;
        FORMAT SESTAT F10.2;
        KEEP &BYVAR COVPARM STAT SESTAT;
RUN;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE COV_TEMP  BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5 MEAN_
TEMP BRRDATA;
RUN;

/* INTRACLASS CORRELATION */

PROC SORT DATA=BRR_TEMP_VAR ;
        BY &BYVAR L COVPARM;
RUN;

PROC TRANSPOSE DATA=BRR_TEMP_VAR OUT=BRR_TEMP0;
        VAR ESTIMATE;
        BY &BYVAR L;
        ID COVPARM;
RUN;

DATA BRR_TEMP1;
        SET BRR_TEMP0;
        ESTIMATE=(INTERCEPT/(INTERCEPT+RESIDUAL));
        KEEP &BYVAR L ESTIMATE;
RUN;

DATA BRR_TEMP2(KEEP=&BYVAR ESTIMATE) BRR_TEMP3(KEEP=&BYVAR STAT);
        SET BRR_TEMP1;
        IF L > 0 THEN OUTPUT BRR_TEMP2;
        ELSE DO;
                STAT =ESTIMATE;
                OUTPUT BRR_TEMP3;
        END;
RUN;

PROC SORT DATA=BRR_TEMP2;
        BY &BYVAR;
RUN;

PROC SORT DATA=BRR_TEMP3;
        BY &BYVAR ;
RUN;

DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY &BYVAR ;
        VARI=((ESTIMATE-STAT)**2)*(1/20);
RUN;

PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
        VAR VARI;
        BY &BYVAR ;
        OUTPUT OUT=BRR_TEMP5 SUM=SS;
RUN;
```


PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 17.16 [5/5] **SAS® macro of PROC_MIXED_NO_PV.sas**

```
DATA &OUTFILE._INTRACLASS;
        MERGE BRR_TEMP3 BRR_TEMP5;
        BY &BYVAR ;
        SESTAT=(SS)**0.5;
        FORMAT STAT F10.2;
        FORMAT SESTAT F10.2;
        KEEP &BYVAR STAT SESTAT;
RUN;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE COV_TEMP  BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5 MEAN_
TEMP BRRDATA;
RUN;


PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP_VAR COV_TEMP  BRR_TEMP0 BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_
TEMP4 BRR_TEMP5 MEAN_TEMP BRRDATA;
RUN;


OPTIONS NOTES;

%MEND;
```


PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 17.17 [1/6] **SAS® macro of PROC_MIXED_PV.sas**

```
%macro BRR_MIXED_PV(INFILE=,
                    REPLI_ROOT=,
                    PV_ROOT=,
                    FIXEF=,
                    RANEF=,
                    BYVAR=,
                    LEVEL2=,
                    OUTSCREEN=,
                    OUTFILE=,);

/*

MEANING OF THE MACRO ARGUMENTS

INFILE = INPUT DATA FILE.
REPLI_ROOT = ROOT OF THE FINAL WEIGHT AND 80 REPLICATES VARIABLE NAMES. FINAL
WEIGHT VARIABLE NAME MUST BE THE REPLICATION ROOT FOLLOWED BY 0.
PV_ROOT = ROOT OF THE 5 PROFICIENCY LEVEL VARIABLES NAMES
FIXEF = LIST OF INDEPENDENT VARIABLES WITH FIXED EFFECT
RANEF = LIST OF INDEPENDENT VARIABLES WITH RANDOM EFFECT
LEVEL2 = LEVEL 2 IDENTIFICATION VARIABLE
BYVAR = BREAKDOWN VARIABLES
OUTSCREEN = DOS ADRESS FOR EXPORTING LISTING
OUTFILE = FILE WITH THE STATISTIC ESTIMATES AND THEIR STANDARD ERROR ESTIMATES.

*/

OPTION NONOTES CLEANUP;

FILENAME MYOUTPUT &OUTSCREEN;
PROC PRINTTO PRINT=MYOUTPUT NEW;
RUN;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP0;
RUN;

PROC SORT DATA=&INFILE
        OUT=BRR_PREP1(KEEP=&REPLI_ROOT.0-&REPLI_ROOT.80  &PV_ROOT.1-&PV_ROOT.5
&BYVAR &FIXEF &RANEF &LEVEL2);
        BY &BYVAR;
RUN;

%LET I=1;
%DO %WHILE(%LENGTH(%SCAN(&FIXEF,&I)));
        %LET I=%EVAL(&I+1);
%END;
%LET NB_FIXE=%EVAL(&I-1);


%LET I=1;
%DO %WHILE(%LENGTH(%SCAN(&RANEF,&I)));
        %LET I=%EVAL(&I+1);
%END;
%LET NB_RAN=%EVAL(&I-1);
%LET NB_TOT=%EVAL(&NB_FIXE+&NB_RAN+5);

DATA BRR_PREP2;
                SET BRR_PREP1;
                NB_MISS=0;
                ARRAY LIST_VAR (&NB_TOT) &PV_ROOT.1-&PV_ROOT.5 &FIXEF &RANEF;
                DO K=1 TO &NB_TOT;
                        IF (LIST_VAR(K) IN (.,.I,.M,.N)) THEN NB_MISS=NB_MISS+1;
                END;
                IF (NB_MISS>1) THEN NB_MISS=1;
RUN;

PROC FREQ DATA=BRR_PREP2 NOPRINT;
        TABLE NB_MISS/OUT=&OUTFILE._DELETION;
        BY &BYVAR;
        WEIGHT &REPLI_ROOT.0;
RUN;
```


PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 17.17 [2/6] **SAS® macro of PROC_MIXED_PV.sas**

```
PROC UNIVARIATE DATA=BRR_PREP2 NOPRINT;
        VAR &REPLI_ROOT.0;
        BY &BYVAR;
        WHERE (NB_MISS=0);
        OUTPUT OUT=BRR_PREP3 N=NBRE SUM=SOMWGT;
RUN;

DATA BRRDATA;
        MERGE BRR_PREP2 BRR_PREP3;
        BY &BYVAR;
        ARRAY OLDWGT (81)&REPLI_ROOT.0-&REPLI_ROOT.80;
    ARRAY NEWWGT (81) STD_WGT0-STD_WGT80;
        DO I=1 TO 81;
                NEWWGT(I)=(OLDWGT(I)/SOMWGT)*NBRE;
        END;
RUN;

%DO I = 1 %TO 5;

        %DO J = 0 %TO 80;

                PROC MIXED DATA=BRRDATA NOCLPRINT NOITPRINT NOINFO METHOD=ML;
                        CLASS &LEVEL2;
                        MODEL &PV_ROOT&I=&FIXEF &RANEF /SOLUTION;
                        RANDOM INTERCEPT &RANEF /SUBJECT=&LEVEL2;
                        WEIGHT STD_WGT&J;
                        BY &BYVAR;
                        ODS OUTPUT SOLUTIONF=FIXE_TEMP COVPARMS=COV_TEMP;
                RUN;

                DATA FIXE_TEMP;
                        SET FIXE_TEMP;
                        L=&J;
                        PV&I=ESTIMATE;
                        KEEP &BYVAR L EFFECT PV&I ;
                RUN;
                PROC SORT DATA=FIXE_TEMP;
                        by &BYVAR L EFFECT;
                RUN;
                DATA COV_TEMP;
                        SET COV_TEMP;
                        L=&J;
                        PV&I=ESTIMATE;
                        KEEP &BYVAR L COVPARM PV&I ;
                RUN;
                PROC SORT DATA=COV_TEMP;
                        BY &BYVAR L COVPARM;
                RUN;

                %IF (&I=1 AND &J=0) %then %do;
                        DATA BRR_TEMP_FIXE;
                                SET FIXE_TEMP;
                        RUN;
                        DATA BRR_TEMP_VAR;
                                SET COV_TEMP;
                        RUN;
                %END;

                DATA BRR_TEMP_FIXE;
                        MERGE BRR_TEMP_FIXE FIXE_TEMP;
                        BY &BYVAR L EFFECT;
                RUN;
                DATA BRR_TEMP_VAR;
                        MERGE BRR_TEMP_VAR COV_TEMP;
                        BY &BYVAR L COVPARM;
                RUN;

        %END;

%END;
```


PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 17.17 [3/6] **SAS® macro of PROC_MIXED_PV.sas**

```
PROC SORT DATA=BRR_TEMP_FIXE OUT=BRR_TEMP1;
        BY &BYVAR L EFFECT;
RUN;

DATA BRR_TEMP2(DROP=STAT FIN1-FIN5 MESVAR)BRR_TEMP3(KEEP=&BYVAR EFFECT STAT FIN1-
FIN5 MESVAR);
        SET BRR_TEMP1;
        IF L > 0 THEN OUTPUT BRR_TEMP2;
        ELSE DO;
                STAT =(PV1+PV2+PV3+PV4+PV5)/5;
                FIN1=PV1;
                FIN2=PV2;
                FIN3=PV3;
                FIN4=PV4;
                FIN5=PV5;
                MESVAR=(((STAT-FIN1)**2)+((STAT-FIN2)**2)+((STAT-FIN3)**2)+((STAT-
FIN4)**2)+((STAT-FIN5)**2))/4;
                OUTPUT BRR_TEMP3;
        END;
RUN;

PROC SORT DATA=BRR_TEMP2;
        BY &BYVAR EFFECT L;
RUN;

PROC SORT DATA=BRR_TEMP3;
        BY &BYVAR EFFECT;
RUN;

DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY &BYVAR EFFECT;
        ARRAY A (5)
                PV1-PV5;
        ARRAY B (5)
                FIN1-FIN5;
        ARRAY C(5)
                VAR1-VAR5;
        DO I=1 TO 5;
                C(I)=(1/20)*((A(I)-B(I))**2);
        END;
RUN;

PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
        VAR VAR1 VAR2 VAR3 VAR4 VAR5;
        BY &BYVAR EFFECT;
        OUTPUT OUT=BRR_TEMP5 SUM=SS1 SS2 SS3 SS4 SS5;
RUN;

DATA &OUTFILE._FIXE;
        MERGE BRR_TEMP3 BRR_TEMP5;
        BY &BYVAR EFFECT;
        SAMP=(SS1+SS2+SS3+SS4+SS5)/5;
        FINVAR=(SAMP+(1.2*MESVAR));
        SESTAT=(FINVAR)**0.5;
        FORMAT STAT F10.1;
        FORMAT SESTAT F10.2;
        KEEP &BYVAR EFFECT STAT SESTAT;
RUN;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP_FIXE FIXE_TEMP BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4
BRR_TEMP5 MEAN_TEMP BRRDATA;
RUN;

PROC SORT DATA=BRR_TEMP_VAR OUT=BRR_TEMP1;
        BY &BYVAR L COVPARM;
RUN;
```

Box 17.17 [4/6] **SAS® macro of PROC_MIXED_PV.sas**

```
DATA BRR_TEMP2(DROP=STAT FIN1-FIN5 MESVAR)BRR_TEMP3(KEEP=&BYVAR COVPARM STAT FIN1-
FIN5 MESVAR);
        SET BRR_TEMP1;
        IF L > 0 THEN OUTPUT BRR_TEMP2;
        ELSE DO;
                STAT =(PV1+PV2+PV3+PV4+PV5)/5;
                FIN1=PV1;
                FIN2=PV2;
                FIN3=PV3;
                FIN4=PV4;
                FIN5=PV5;
                MESVAR=(((STAT-FIN1)**2)+((STAT-FIN2)**2)+((STAT-FIN3)**2)+((STAT-
FIN4)**2)+((STAT-FIN5)**2))/4;
                OUTPUT BRR_TEMP3;
        END;
RUN;

PROC SORT DATA=BRR_TEMP2;
        BY &BYVAR COVPARM L;
RUN;

PROC SORT DATA=BRR_TEMP3;
        BY &BYVAR COVPARM;
RUN;

DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY &BYVAR COVPARM;
        ARRAY A (5)
                PV1-PV5;
        ARRAY B (5)
                FIN1-FIN5;
        ARRAY C(5)
                VAR1-VAR5;
        DO I=1 TO 5;
                C(I)=(1/20)*((A(I)-B(I))**2);
        END;
RUN;

PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
        VAR VAR1 VAR2 VAR3 VAR4 VAR5;
        BY &BYVAR COVPARM;
        OUTPUT OUT=BRR_TEMP5 SUM=SS1 SS2 SS3 SS4 SS5;
RUN;

DATA &OUTFILE._VARIANCE;
        MERGE BRR_TEMP3 BRR_TEMP5;
        BY &BYVAR COVPARM;
        SAMP=(SS1+SS2+SS3+SS4+SS5)/5;
        FINVAR=(SAMP+(1.2*MESVAR));
        SESTAT=(FINVAR)**0.5;
        FORMAT STAT F10.1;
        FORMAT SESTAT F10.2;
        KEEP &BYVAR COVPARM STAT SESTAT;
RUN;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE  BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_TEMP4 BRR_TEMP5 MEAN_TEMP
/*BRRDATA*/;
RUN;

PROC SORT DATA=BRR_TEMP_VAR ;
        BY &BYVAR L COVPARM;
RUN;

%DO K=1 %TO 5;

PROC TRANSPOSE DATA=BRR_TEMP_VAR OUT=BRR_TEMP0;
        VAR PV&K;
        BY &BYVAR L;
        ID COVPARM;
RUN;
```


PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 17.17 [5/6] **SAS® macro of PROC_MIXED_PV.sas**

```
DATA BRR_TEMP0;
        SET BRR_TEMP0;
        PV&K=(INTERCEPT/(INTERCEPT+RESIDUAL));
        KEEP &BYVAR L PV&K;
RUN;

%IF (&K=1) %THEN %DO;
        DATA BRR_TEMP1;
                SET BRR_TEMP0;
        RUN;
%END;
DATA BRR_TEMP1;
        MERGE BRR_TEMP1 BRR_TEMP0;
        BY &BYVAR L;
RUN;

%END;


DATA BRR_TEMP2(DROP=STAT FIN1-FIN5 MESVAR)BRR_TEMP3(KEEP=&BYVAR STAT FIN1-FIN5
MESVAR);
        SET BRR_TEMP1;
        IF L > 0 THEN OUTPUT BRR_TEMP2;
        ELSE DO;
                STAT =(PV1+PV2+PV3+PV4+PV5)/5;
                FIN1=PV1;
                FIN2=PV2;
                FIN3=PV3;
                FIN4=PV4;
                FIN5=PV5;
                MESVAR=(((STAT-FIN1)**2)+((STAT-FIN2)**2)+((STAT-FIN3)**2)+((STAT-
FIN4)**2)+((STAT-FIN5)**2))/4;
                OUTPUT BRR_TEMP3;
        END;
RUN;

PROC SORT DATA=BRR_TEMP2;
        BY &BYVAR  L;
RUN;

PROC SORT DATA=BRR_TEMP3;
        BY &BYVAR ;
RUN;

DATA BRR_TEMP4;
        MERGE BRR_TEMP2 BRR_TEMP3;
        BY &BYVAR ;
        ARRAY A (5)
                PV1-PV5;
        ARRAY B (5)
                FIN1-FIN5;
        ARRAY C(5)
                VAR1-VAR5;
        DO I=1 TO 5;
                C(I)=(1/20)*((A(I)-B(I))**2);
        END;
RUN;

PROC UNIVARIATE DATA=BRR_TEMP4 NOPRINT;
        VAR VAR1 VAR2 VAR3 VAR4 VAR5;
        BY &BYVAR ;
        OUTPUT OUT=BRR_TEMP5 SUM=SS1 SS2 SS3 SS4 SS5;
RUN;
```

Box 17.17 [6/6] **SAS® macro of PROC_MIXED_PV.sas**

```
DATA &OUTFILE._INTRACLASS;
        MERGE BRR_TEMP3 BRR_TEMP5;
        BY &BYVAR ;
        SAMP=(SS1+SS2+SS3+SS4+SS5)/5;
        FINVAR=(SAMP+(1.2*MESVAR));
        SESTAT=(FINVAR)**0.5;
        FORMAT STAT F10.2;
        FORMAT SESTAT F10.2;
        KEEP &BYVAR STAT SESTAT;
RUN;

PROC DATASETS LIBRARY=WORK NOLIST;
        DELETE BRR_TEMP_VAR COV_TEMP  BRR_TEMP0 BRR_TEMP1 BRR_TEMP2 BRR_TEMP3 BRR_
TEMP4 BRR_TEMP5 MEAN_TEMP
        BRR_PREP1 BRR_PREP2 BRR_PREP3;
RUN;

PROC PRINTTO PRINT=PRINT;
RUN;

OPTIONS NOTES;

%MEND;
```

References

Beaton, A.E. (1987), *The NAEP 1983-1984 Technical Report*, Educational Testing Service, Princeton.

Beaton, A.E., *et al.* (1996), *Mathematics Achievement in the Middle School Years, IEA's Third International Mathematics and Science Study*, Boston College, Chestnut Hill, MA.

Bloom, B.S. (1979), *Caractéristiques individuelles et apprentissage scolaire*, Éditions Labor, Brussels.

Bressoux, P. (2008), *Modélisation statistique appliquée aux sciences sociales*, De Boek, Brussels.

Bryk, A.S. and **S.W. Raudenbush** (1992), *Hierarchical Linear Models for Social and Behavioural Research: Applications and Data Analysis Methods*, Sage Publications, Newbury Park, CA.

Buchmann, C. (2000), *Family structure, parental perceptions and child labor in Kenya: What factors determine who is enrolled in school?* a*Soc. Forces*, No. 78, pp. 1349-79.

Cochran, W.G. (1977), *Sampling Techniques*, J. Wiley and Sons, Inc., New York.

Dunn, O.J. (1961), "Multilple Comparisons among Menas", *Journal of the American Statistical Association*, Vol. 56, American Statistical Association, Alexandria, pp. 52-64.

Kish, L. (1995), *Survey Sampling*, J. Wiley and Sons, Inc., New York.

Knighton, T. and **P. Bussière** (2006), "Educational Outcomes at Age 19 Associated with Reading Ability at Age 15", Statistics Canada, Ottawa.

Gonzalez, E. and **A. Kennedy** (2003), *PIRLS 2001 User Guide for the International Database*, Boston College, Chestnut Hill, MA.

Ganzeboom, H.B.G., P.M. De Graaf and **D.J. Treiman** (1992), "A Standard International Socio-economic Index of Occupation Status", *Social Science Research* 21(1), Elsevier Ltd, pp 1-56.

Goldstein, H. (1995), *Multilevel Statistical Models*, 2nd Edition, Edward Arnold, London.

Goldstein, H. (1997), "Methods in School Effectiveness Research", *School Effectiveness and School Improvement* 8, Swets and Zeitlinger, Lisse, Netherlands, pp. 369-395.

Hubin, J.P. (ed.) (2007), *Les indicateurs de l'enseignement*, 2nd Edition, Ministère de la Communauté française, Brussels.

Husen, T. (1967), *International Study of Achievement in Mathematics: A Comparison of Twelve Countries*, Almqvist and Wiksells, Uppsala.

International Labour Organisation (ILO) (1990), *International Standard Classification of Occupations: ISCO-88.* Geneva: International Labour Office.

Lafontaine, D. and **C. Monseur** (forthcoming), "Impact of Test Characteristics on Gender Equity Indicators in the Assessment of Reading Comprehension", *European Educational Research Journal*, Special Issue on PISA and Gender.

Lietz, P. (2006), "A Meta-Analysis of Gender Differences in Reading Achievement at the Secondary Level", *Studies in Educational Evaluation* 32, pp. 317-344.

Monseur, C. and **M. Crahay** (forthcoming), "Composition académique et sociale des établissements, efficacité et inégalités scolaires : une comparaison internationale – Analyse secondaire des données PISA 2006", *Revue française de pédagogie.*

OECD (1998), *Education at a Glance – OECD Indicators*, OECD, Paris.

OECD (1999a), *Measuring Student Knowledge and Skills – A New Framework for Assessment*, OECD, Paris.

OECD (1999b), *Classifying Educational Programmes – Manual for ISCED-97 Implementation in OECD Countries*, OECD, Paris.

OECD (2001), *Knowledge and Skills for Life – First Results from PISA 2000*, OECD, Paris.

OECD (2002a), *Programme for International Student Assessment – Manual for the PISA 2000 Database*, OECD, Paris.

OECD (2002b), *Sample Tasks from the PISA 2000 Assessment – Reading, Mathematical and Scientific Literacy,* OECD, Paris.

OECD (2002c), *Programme for International Student Assessment – PISA 2000 Technical Report,* OECD, Paris.

OECD (2002d), *Reading for Change: Performance and Engagement across Countries – Results from PISA 2000,* OECD, Paris.

OECD (2003a), *Literacy Skills for the World of Tomorrow – Further Results from PISA 2000,* OECD, Paris.

OECD (2003b), *The PISA 2003 Assessment Framework – Mathematics, Reading, Science and Problem Solving Knowledge and Skills,* OECD, Paris.

OECD (2004a), *Learning for Tomorrow's World – First Results from PISA 2003,* OECD, Paris.

OECD (2004b), *Problem Solving for Tomorrow's World – First Measures of Cross-Curricular Competencies from PISA 2003,* OECD, Paris.

OECD (2005a), *PISA 2003 Technical Report,* OECD, Paris.

OECD (2005b), *PISA 2003 Data Analysis Manual,* OECD, Paris.

OECD (2006), *Assessing Scientific, Reading and Mathematical Literacy: A Framework for PISA 2006,* OECD, Paris.

OECD (2007), *PISA 2006: Science Competencies for Tomorrow's World,* OECD, Paris.

OECD (2009), *PISA 2006 Technical Report,* OECD, Paris.

Peaker, G.F. (1975), *An Empirical Study of Education in Twenty-One Countries: A Technical report. International Studies in Evaluation VIII,* Wiley, New York and Almqvist and Wiksell, Stockholm.

Rust, K.F. and **J.N.K. Rao** (1996), "Variance Estimation for Complex Surveys Using Replication Techniques", *Statistical Methods in Medical Research,* Vol. 5, Hodder Arnold, London, pp. 283-310.

Rutter, M., *et al.* (2004), "Gender Differences in Reading Difficulties: Findings from Four Epidemiology Studies", *Journal of the American Medical Association* 291, pp. 2007-2012.

Schulz, W. (2006), *Measuring the socio-economic background of students and its effect on achievement in PISA 2000 and PISA 2003,* Paper presented at the Annual Meetings of the American Educational Research Association (AERA) in San Francisco, 7-11 April.

Wagemaker, H. (1996), *Are Girls Better Readers. Gender Differences in Reading Literacy in 32 Countries,* IEA, The Hague.

Warm, T.A. (1989), "Weighted Likelihood Estimation of Ability in Item Response Theory", *Psychometrika,* Vol. 54(3), Psychometric Society, Williamsburg, VA., pp. 427-450.

Wright, B.D. and **M.H. Stone** (1979), *Best Test Design: Rasch Measurement,* MESA Press, Chicago.

Table of contents

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

LIST OF BOXES

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

LIST OF FIGURES

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

LIST OF TABLES

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

User's Guide

Preparation of data files

All data files (in text format) and the SAS® control files are available on the PISA website (*www.pisa.oecd.org*).

SAS® users

By running the SAS® control files, the PISA data files are created in the SAS® format. Before starting analysis, assigning the folder in which the data files are saved as a SAS® library.

For example, if the PISA 2000 data files are saved in the folder of "c:\pisa2000\data\", the PISA 2003 data files are in "c:\pisa2003\data\", and the PISA 2006 data files are in "c:\pisa2006\data\", the following commands need to be run to create SAS® libraries:

```
libname PISA2000 "c:\pisa2000\data\";
libname PISA2003 "c:\pisa2003\data\";
libname PISA2006 "c:\pisa2006\data\";
run;
```

SAS® syntax and macros

All syntaxes and macros in this manual can be copied from the PISA website (*www.pisa.oecd.org*). The 17 SAS® macros presented in Chapter 17 need to be saved under "c:\pisa\macro\", before staring analysis. Each chapter of the manual contains a complete set of syntaxes, which must be done sequentially, for all of them to run correctly, within the chapter.

Rounding of figures

In the tables and formulas, figures were rounded to a convenient number of decimal places, although calculations were always made with the full number of decimal places.

Country abbreviations used in this manual

AUS	Australia	FRA	France	MEX	Mexico
AUT	Austria	GBR	United Kingdom	NLD	Netherlands
BEL	Belgium	GRC	Greece	NOR	Norway
CAN	Canada	HUN	Hungary	NZL	New Zealand
CHE	Switzerland	IRL	Ireland	POL	Poland
CZE	Czech Republic	ISL	Iceland	PRT	Portugal
DEU	Germany	ITA	Italy	SVK	Slovak Republic
DNK	Denmark	JPN	Japan	SWE	Sweden
ESP	Spain	KOR	Korea	TUR	Turkey
FIN	Finland	LUX	Luxembourg	USA	United States

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009